

LOGO

MICRODIGITAL

LOGO
para TK Color Computer

Projeto e execução editorial: Departamento de Publicações Técnicas
Microdigital Eletrônica Ltda.

Copirraite c 1986 de Microdigital Eletrônica Ltda - Todos os direitos reservados.

É vetada a reprodução total ou parcial deste manual sem prévia autorização por escrito da Microdigital Eletrônica Ltda.

A Microdigital reserva-se o direito de alterar qualquer especificação deste produto futuramente.

MICRODIGITAL ELETRÔNICA LTDA

*Caixa Postal 54.121
São Paulo (SP)*

*Impresso no Brasil - Printed in Brazil
1.^a edição - julho 1986.*

NOTA:

Este livro originalmente possui muitas páginas em branco que foram retiradas deste arquivo para que ficasse melhor a leitura, a visualização e o tamanho deste mesmo arquivo.

SUMÁRIO

Aos Professores	5
APRESENTAÇÃO	7
CARREGANDO O LOGO EM SEU COMPUTADOR.....	7
1 - PRIMEIROS CONTATOS.....	11
2 - FAZENDO TRAÇOS.....	19
3 - DESENHANDO.....	25
4 - ENSINANDO A TARTARUGA.....	35
5 - VARIÁVEIS E LISTAS	45
6 - ESCRREVENDO.....	55
7 - REVENDO O QUE A TARTARUGA APRENDEU E GUARDANDO EM FITA	61
8 - ESTRUTURANDO	71
9 - OPERAÇÕES ARITMÉTICAS E COMPARAÇÕES	81
10 - DESENHANDO UMA CASA COM JARDIM.....	93
11 - O ESPAÇO DA TARTARUGA.....	101
12 - OUTRAS FUNÇÕES MATEMÁTICAS	109
13 - PALAVRAS E LISTAS.....	115
14 - ENTRADAS E SAÍDAS.....	129
15 - OUTROS COMANDOS E OPERAÇÕES.....	135

Aos Professores:

O LOGO é uma linguagem de programação cujo objetivo é servir de instrumental no ensino de crianças de sete a doze anos. Experiências realizadas nos Estados Unidos e na Europa demonstraram que os princípios que norteiam o LOGO podem ser aprendidos por crianças de até três anos de idade.

Não se trata de uma linguagem semelhante às que normalmente são usadas em processamento de dados, embora possa ser usada para este fim. As crianças não fazem programas; ensinam a “tartaruga”, um simpático bichinho que reside na tela do computador. A programação é aprendida (e apreendida) intuitivamente. O mesmo processo dá-se com outros conceitos, principalmente os matemáticos.

O método de ensino empregado deve ser o mais livre possível, dando-se ênfase ao aspecto lúdico, que conduz a um aprendizado intuitivo.

Nada deve ser forçado. Os erros devem ser encarados como parte do processo de aprendizagem. Não deve haver qualquer tipo de avaliação classificatória (por exemplo, notas). A criança deve ser estimulada a procurar seus problemas (chamados de “grilos”) e resolvê-los.

O computador não deve ser encarado como um substituto do professor, mas como um instrumento didático, bem mais poderoso que os costumeiros. A propósito, a grande vantagem deste recurso didático está no fato de não ser ele do tipo “mão-única”, possibilitando uma interação entre a criança e o professor.

O melhor que se tem a fazer é encarar o Logo como uma “massa de modelar lógica” com a qual cada criança cria, com toda a liberdade, os seus projetos.

Finalmente é necessário assinalar que o Logo não é uma panacéia universal, um recurso definitivo em matéria de aprendizagem. A escola pode passar por grandes transformações com a chegada do computador, porém ele sempre será um instrumento de trabalho que depende de “pessoas” para que sirva a objetivos positivos.

APRESENTAÇÃO

Você gostaria de ter um bichinho que fizesse tudo o que você mandasse?

Assim que você carregar a fita ou o cartucho do Logo, vai ver, na tela do computador, uma simpática tartaruguinha. Muito obediente e inteligente. Você vai poder ensinar a ela uma porção de coisas novas. Ela vai aprender rapidamente.

Para poder conversar com ela, você precisa aprender o Logo. O Logo é a linguagem que a tartaruguinha entende. Não é uma linguagem difícil de ser aprendida. Você vai achar fácil e divertido falar em Logo com a sua tartaruguinha.

Juntos, você e ela vão fazer coisas maravilhosas. Desenhos coloridos, jogos de palavras, contas, cartas e muito mais.

Veja a seguir como carregar o programa no computador, e mãos à obra!

CARREGANDO O LOGO EM SEU COMPUTADOR

Para que o seu computador “entenda” o Logo, é necessário que um intérprete faça a tradução desta linguagem para a linguagem que ele entende. O intérprete é um programa especial que está gravado na fita ou cartucho que você tem à disposição.

O que você necessita:

- 1 - O seu computador TK com 48 K*
- 2 - Um gravador*
- 3 - A fita ou cartucho com LOGO*
- 4 - Um fita virgem*
- 5 - este manual*
- 6 - TV colorido ou B & P*

FITA

Se você possui uma fita, a carga deve ser feita da seguinte forma:

- 1) Veja se o computador possui uma ou duas entradas para o gravador cassete.*

Se possui apenas uma, ligue o cabo para gravação na entrada CASSETE do computador e na entrada EAR do gravador.

Se possui duas entradas, ligue o cabo apropriado na tomada EAR do computador e na tomada EAR do gravador.

- 2) Ajuste o volume do gravador numa altura média. Este ajuste não precisa ser exato.
- 3) Coloque a fita no começo (qualquer um dos lados pode ser usado, pois o programa está gravado em ambos).
- 4) Ligue a TV e o computador (coloque a chave da fonte de alimentação em ON). Coloque a TV no canal 3. Ajuste a sintonia fina até que a imagem fique boa (verifique no manual da TV como fazer isso).
- 5) Aperte a tecla J uma vez. Em seguida, mantendo a tecla SYMBOL SHIFT abaixada, aperte P ou apenas “ duas vezes. Você deverá ver na tela:

LOAD “”

- 6) Aperte a tecla ENTER.
- 7) A carga começará a ser feita. Na tela você verá um quadrado branco com o nome do programa e um série de listas coloridas. Em seguida, verá a formação de uma imagem. Depois de alguns minutos, você verá a mensagem:

LOGO
TK COLOR COMPUTER 1986
?

- 8) Desligue o gravador. Retire a fita e guarde-a no estojo. Ponha uma fita nova para, mais tarde, gravar o que você desejar.

Se tudo correu bem até aqui, você já pode ir ao primeiro capítulo.

Se alguma coisa saiu errado, faça tudo outra vez. Se depois de duas ou três tentativas você não conseguir, consulte o manual do seu computador. Ele contém várias informações sobre como carregar programas. Leia-o e tente descobrir onde foi que você errou, para não ter os mesmos problemas da próxima vez.

CARTUCHO

Se você possui um cartucho, o processo de carga é simples. Basta deixar o

*computador desligado e conectar o cartucho na saída **EXPANSION** do computador.*

Quando você ligar o computador, o Logo entrará automaticamente, mostrando a mensagem:

LOGO
TK COLOR COMPUTER 1986
?

Pronto, você já pode ir ao capítulo 1.

PRIMEIROS

CONTATOS

1

1. PRIMEIROS CONTATOS

Como você e a tartaruga comunicam-se?

Vocês já entraram em contato. Você carregou o intérprete e a tartaruga deu-lhe as boas-vindas.

O meio de comunicação da tartaruga com você é a tela da televisão. Algumas vezes ela poderá usar o alto-falante, mas, em geral, usará a tela.

Por enquanto ela ainda não está visível, está esperando suas instruções.

Na tela você vê a mensagem, um ponto de interrogação e um quadrado preto que pisca sem parar. O ponto de interrogação indica que você pode mandar a tartaruga fazer alguma coisa.

LOGO
TK COLOR COMPUTER 1986
?

O seu meio de comunicação com a tartaruga é o teclado. Tudo o que você quiser dizer à tartaruga deve ser escrito, letra por letra, pelo teclado.

Vamos chamar a tartaruga?

A palavra em Logo que diz para a tartaruga aparecer é VENHA. Mas, será que é suficiente escrever VENHA, para que ela entenda?

Vamos experimentar:

1) *Escreva venha, letra por letra.*

Escrever alguma coisa letra por letra chama-se “digitar”.

2) *Aperte a tecla ENTER. Toda vez que você quiser que a tartaruga faça alguma coisa, aperte ENTER depois de ter digitado a sua ordem.*

A tartaruga surgiu na tela? Se você digitou certo, ela deve ter aparecido no centro da tela. Se você cometeu algum erro, por exemplo se esqueceu a letra a, a tartaruguinha responderá:

AINDA NAO APRENDI VENH


Ela responderá assim sempre que você pedir-lhe para fazer algo que ela ainda não aprendeu. Mais tarde você saberá como ensinar palavras novas à sua tartaruga.

Telas e Teclas

A tela pode ser entendida como um papel onde você e a tartaruga vão escrever ou desenhar.


Quando o programa Logo é carregado, a tela fica preparada para receber letras, por isso a tartaruga não aparece. Ela só vai aparecer quando você quiser fazer desenhos.

Quando você chama a tartaruga, através do comando VENHA, a tela fica preparada para desenhos. Apenas duas linhas, no final da tela, podem ser usadas para letras.



Para voltar à tela de letras, use o comando TELATEXTO. Digite TELATEXTO e aperte a tecla ENTER.

Tudo o que estiver na tela desaparecerá, e o ponto de interrogação aparecerá no alto, à esquerda, indicando que você pode escrever na tela.



Note que, tanto na tela de letras como na tela de desenhos, aparece uma letra L no canto inferior direito. Esta letra indica que as letras minúsculas podem ser digitadas. Chamamos isso de “modo L”.

Quando a letra L está sendo mostrada, as letras maiúsculas são conseguidas pela tecla CAPS SHIFT mais a tecla da letra desejada.

A tartaruga não faz diferença entre as letras minúsculas e as maiúsculas. Neste livro usaremos as letras maiúsculas, para realçar as ordens. Mas, você pode usar apenas as minúsculas.

Se você preferir digitar suas ordens com letras maiúsculas, aperte CAPS SHIFT e 2 ao mesmo tempo ou <CAPS LOCK>. A letra do canto inferior esquerdo mudará de L para C. Tudo o que você digitar será escrito com

letras maiúsculas. Para escrever novamente com letras minúsculas, pressione novamente *CAPS SHIFT 2 <CAPS LOCK>*. A letra L reaparecerá e tudo volta a ser como antes.

Os símbolos marcados em vermelho no teclado (por ex., = ? + >) são conseguidos quando você aperta, ao mesmo tempo, *SYMBOL SHIFT* e a tecla do símbolo desejado.

Sempre que dissermos “aperte *CAPS SHIFT* ou *SYMBOL SHIFT* e alguma tecla, ao mesmo tempo”, significa que você deve manter *CAPS* ou *SYMBOL SHIFT* abaixada enquanto aperta outra tecla.

Cursor

O cursor é aquele quadradinho preto que não pára de piscar. Ele acompanha cada letra ou algarismo que você digita.

Você pode usar o cursor para apagar as letras e números que digitou errado.

Para fazer o cursor apagar alguma coisa que você digitou errado, aperte *CAPS SHIFT* e 0 ao mesmo tempo (*DELETE*).

O cursor apagará a última letra digitada. Agora você poderá escrever a letra certa.

Por exemplo, se você digitar *TELATEXTO* da seguinte forma:

?TELATEXTP

(Não aperte *ENTER* ainda.)

Quando você apertar *CAPS SHIFT* e 0 ao mesmo tempo (*DELETE*), o cursor apagará a letra P, ficando no lugar dela.

?TELATEXT

Agora você pode digitar a letra O e apertar *ENTER*:

?TELATEXTO <ENTER>

As palavras entre < > indicam teclas a serem apertadas, e não palavras a serem digitadas. Você não deve digitar os símbolos < > nem as letras que estiverem entre eles.

Você também pode mover o cursor sem apagar nada. Se você quiser colocar o cursor sobre uma letra à esquerda, sem apagá-la, aperte *CAPS SHIFT* e 5 <←>.

Vamos simplificar? Daqui em diante, em vez de falarmos CAPS SHIFT, falaremos CAPS e, em vez de SYMBOL SHIFT, falaremos SYMBOL.

O cursor move-se para a letra à esquerda mostrando-a em branco sobre fundo preto. Ele continua piscando, mostrando a letra ora em preto sobre fundo branco, ora em branco sobre fundo preto.

Você pode digitar agora quantas letras ou números desejar. Se você apertar ENTER, a palavra inteira será considerada. Tanto o que estiver antes como o que estiver depois do cursor.

Por exemplo, se em vez de digitar TELATEXTO, você digitar:

?TELATXTO

Aperte CAPS e 5 <←> três vezes:

?TELATXTO <CAPS 5> <←>
?TELATXT <CAPS 5> <←>
?TELATX O <CAPS 5> <←>
?TELAT X TO

Digite agora a letra E:

?TELATE X TO

O cursor também pode ser movido uma letra para a direita. Para isso, use CAPS e 8 <→>. Se você quiser mudar o cursor que está sobre a letra X na palavra TELATEXTO, passando-o para depois da letra O, aperte CAPS e 8 <→> três vezes:


?TELATE X TO <CAPS 8> <→>
?TELAT E XTO <CAPS 8> <→>
?TELATEXT O <CAPS 8> <→>
?TELATEXTO

Você sempre pode acrescentar algo ou apagar o que estiver escrito à esquerda do cursor.


Agora você já sabe como se comunicar com a tartaruga. Para que ela possa fazer mais coisas, você deve ir ao próximo capítulo.

Resumo do que você aprendeu neste capítulo:

TELAS DE LETRAS



TELAS DE DESENHOS



<i>COMANDOS</i>	<i>EFEITO</i>
VENHA	<i>Fazer a tartaruga aparecer, para desenhar.</i>
TELATEXTO	<i>Mudar da tela de desenhos para a de letras.</i>

<i>TECLAS</i>	<i>EFEITO</i>
ENTER	<i>Fazer a tartaruga obedecer à ordem.</i>
CAPS SHIFT e LETRA	<i>conseguir letras maiúsculas no “modo L”.</i>
CAPS SHIF 2 <CAPS LOCK>	<i>colocar no “modo C” permitindo que se escreva apenas com letras maiúsculas.</i>
SYMBOL SHIFT e SÍMBOLO	<i>Escrever símbolos marcados em vermelho no teclado. Ex.: = ? > + -</i>
CAPS SHIFT e 0 <DELETE>	<i>Mover o cursor para a esquerda, apagando um letra.</i>
CAPS SHIFT e 5 <←>	<i>Mover o cursor para a esquerda, sem apagar a letra.</i>
CAPS SHIFT e 8 <→>	<i>Mover o cursor para a direita</i>

FAZENDO

TRAÇOS

2

2. FAZENDO TRAÇOS

Uma das coisas que você pode fazer com a tartaruga é desenhar. Ela é muito habilidosa no manejo do lápis e da borracha. Basta que você explique-lhe tudo o que deve ser feito.


As suas explicações precisam ser dadas na linguagem Logo. A linguagem da tartaruga. Você já conhece duas palavras em Logo: VENHA e TELATEXTO. Aos poucos você conhecerá todas as palavras que a tartaruga sabe.

Fazendo a Tartaruga Andar

Vamos chamar a tartaruguinha?

?VENHA

Não se esqueça de apertar <ENTER> ao final de cada comando.




Não adianta muito ter uma tartaruga parada. Vamos fazê-la andar?

Digite:

?AVANCE 50 <ENTER>

A tartaruga andará 50 passos na tela, desenhando um traço pelo seu caminho.



Repare que o número de passos que a tartaruga deve dar está separado por um espaço em branco. Se você se esquecer do espaço, a tartaruga responderá:

AINDA NAO APRENDI AVANCE50


O número de passos é algo que você precisa dizer à tartaruga. É um complemento da ordem AVANCE. Este complemento é chamado de “dado”.

Os dados no Logo precisam ser escritos separados dos comandos. Se você esquecer o espaço, a tartaruga vai achar que é uma palavra nova, e responderá que ainda não a aprendeu.

O número de passos poderá variar de 0 a 32767.

Que tal a tartaruga andar em outra direção? Temos que escolher o lado para o qual ela deve virar, por exemplo, direita.

?DIREITA 90 <ENTER>




A partir de agora, não avisaremos mais quando você deve apertar a tecla ENTER. Toda vez que você quiser que a tartaruga obedeça a um ordem, deve apertar ENTER ao final do comando.

O que significa o número 90 na ordem DIREITA? Significa quantos graus a tartaruga deve virar. Experimente dar várias ordens, trocando o número de graus. Esse número pode variar de 0 a 360. Não se esqueça de colocar um espaço entre a palavra DIREITA e o número.

Mais tarde, explicaremos o que são graus. Você poderá então explorar bastante a capacidade que a tartaruga tem para fazer desenhos.


Será que a tartaruga pode voltar para trás? Digite:

?VOLTE 50




Podemos fazer a tartaruga virar para a esquerda também:

?ESQUERDA 45



Para ver melhor a posição, digite:

?VOLTE 50



Vamos apagar toda a tela? Digite então:

?LIMPETELA

A tartaruga voltará ao centro da tela, e tudo o que você desenhou será apagado.

Simplificando

Até agora fizemos a tartaruga desenhar apenas algumas linhas. E se fossemos desenhar uma estrada com carros e caminhões ou um jardim florido? Imagine como seria difícil e cansativo digitar cada um dos comandos, letra por letra.

Felizmente, no Logo existem muitas formas de simplificar as coisas. Uma delas é escrever apenas algumas letras dos comandos. Isto é chamado “forma abreviada”.

Os comandos mais usados podem ser escritos na forma abreviada.

Todos os comandos que mostramos até aqui possuem uma forma abreviada. Veja na tabela abaixo as abreviaturas que você pode usar no lugar desses comandos:

COMANDOS	ABREVIATURAS
VENHA	V
TELATEXTO	TT
AVANCE	AV
VOLTE	VT
DIREITA	DT
ESQUERDA	EQ
LIMPETELA	LT

Com estes primeiros comandos você já poderá fazer alguns desenhos. Procure fazer algumas experiências.

Não se preocupe com os erros, pois a tartaruga avisará quando você escrever alguma coisa errada. Além disso, sempre é possível limpar a tela e fazer tudo de novo.

Quando se usa a tela de desenhos, a frase escrita nas linhas de letras podem não caber na tela. Então, aparecerá uma flechinha ↑ no lugar da letra L, piscando no canto inferior da tela. Essa flechinha indica que você deve apertar a tecla ENTER, para ver o restante da frase.

No próximo capítulo mostraremos mais alguns comandos para desenhar na tela.

Resumo dos comandos que você aprendeu neste capítulo:

COMANDOS	ABREVIATURA	SIGNIFICADO
AVANCE	AV	Faz com que a tartaruga ande para a frente.
VOLTE	VT	Faz com que a tartaruga ande para trás.
DIREITA	DT	Faz com que a tartaruga vire para a direita.
ESQUERDA	EQ	Faz com que a tartaruga vire para a esquerda.
LIMPETELA	LT	Limpa toda a tela de desenhos e faz a tartaruga voltar ao centro.


DESENHANDO

3

3. DESENHANDO

Para que os seus desenhos caibam na tela, você precisa saber quanto ela mede.


A tela de desenhos mede 256 “passos de tartaruga” no sentido horizontal e 175 no sentido vertical, de uma ponta à outra.



A partir do centro, a tela mede 128 passos na horizontal e 87,5 na vertical.

Experimente mandar a tartaruga andar para além do espaço da tela:

?AVANCE 150




O que aconteceu? A tartaruga, ao encontrar o limite superior da tela, “deu a volta” e surgiu na parte inferior.

Experimente agora os seguintes comandos:

?DT 90 AV 150

Usamos nesta linha dois recursos que o LOGO permite. São eles: o uso da abreviatura dos comandos e o uso de mais de um comando por linha.

Repare no espaço entre os dois comandos. Você pode usar quantos comandos quiser numa linha, desde que deixe pelo menos um espaço para separá-los.



E agora o que aconteceu? A tartaruga “deu a volta” novamente ao encontrar o limite da tela.

Vamos fazer uma experiência? Em primeiro lugar, vamos apagar a tela. O Logo possui dois comandos para isso, as ordens LIMPE e LIMPETELA.

LIMPE apaga todos os desenhos, mas deixa a tartaruga onde ela está. Experimente e veja o que acontece.

LIMPETELA apaga todos os desenhos e leva a tartaruga ao centro da tela. Faça alguns traços e a seguir digite LIMPETELA. Se preferir, use a abreviatura LT.

?LIMPETELA

Vamos aproveitar a capacidade da tartaruga para “dar a volta” e transformar a tela numa toalha de piquenique:

?DT 45 AV 5000 DT 90 AV 5000

Vamos melhorar esta toalha? A tartaruga poderia sair de cima da toalha. Vamos escondê-la?

?ESCONDA

(Se preferir, digite EC)

A tartaruga poderá reaparecer a qualquer momento. Basta você chamá-la novamente:

?VENHA

A tartaruga, mesmo escondida, pode continuar a desenhar. Limpe a tela e esconda a tartaruga:


?LT EC

Experimente agora digitar:

?AV 50 DT 90 AV 50 DT 90 AV 50
DT 90 AV 50 DT 90

Não se importe se os comandos não couberem numa única linha. O Logo passa automaticamente para a linha inferior.

Se você digitou tudo certo, a tartaruga deverá desenhar um quadrado.



Deixe novamente a tartaruga visível, chamando-a por meio do comando VENHA.

Iremos agora explorar outras habilidades artísticas da tartaruga. Em primeiro lugar digite:

?EQ 45 AV 500
?EQ 90 AV 500

(Não se esqueça de apertar ENTER após cada linha.)

Digite agora:

?CENTRO

Quando você dá a ordem CENTRO, a tartaruga vai ao centro da tela, riscando o caminho, e vira para cima. Isso acontece ainda que ela esteja invisível. Faça com a tartaruga andar pela tela. Use várias vezes o comando CENTRO, com a tartaruga em várias posições da tela.

Pronto? Não ficou uma bagunça? Apague a tela, com LT.

Lápis e Borracha

O Logo permite a você disciplinar a tartaruga, evitando que ela desenhe onde você não quer ou, ainda, que ela apague o que não está errado.

Se você não quiser que a tartaruga risque o caminho, deve dizer-lhe para caminhar sem lápis. Digite:

?SEMLAPIS

(Ou, se preferir, a abreviatura SL)
Para que ela volte a fazer traços digite:

?COMLAPIS

(ou, se preferir, a abreviatura CL)

Vamos experimentar:

?EQ 90 AV 50

?EQ 90 AV 50

?EQ 90 AV 50

?EQ 90 AV 50

Você deve ter obtido um quadrado. Vamos colocar a tartaruga dentro deste quadrado, sem fazer riscos?


?SEMLAPIS

?VOLTE 25

?ESQUERDA 90

?AVANCE 25

?DIREITA 90



Peça à tartaruga que use o lápis novamente:

?COMLAPIS

Faça um traço, por exemplo:

?AVANCE 50

Vamos supor que você não queria fazer esse traço? Será que podemos apagar um traço? Digite:

?BORRACHA

(ou simplesmente B)

Este comando faz com que a tartaruga substitua o lápis por uma borracha, apagando qualquer traço, por onde ela passar.

Experimente agora:

?VOLTE 50

O último traço foi apagado!

Para cancelar o efeito da borracha, use os comandos COMLAPIS ou SEMLAPIS. Quando estes comandos forem digitados, nenhum traço será apagado.

O Lápis Mágico

A tartaruga possui um lápis que ninguém mais tem, o lápis invertido.

O lápis invertido risca onde não houver traços e apaga onde já houver traços.

Para que a tartaruga use esse lápis, você comando INVELAPIS ou IL. Vamos experimentar? Digite:

?LIMPETELA
?INVELAPIS
?DT 45 AV 10000

Percebeu o efeito?

Complete agora digitando:

?EQ 90 VT 10000

Para que a tartaruga volte a usar o lápis normal, basta digitar COMLAPIS.

Colorindo

A tartaruga possui vários lápis de cor. As cores que ela pode usar são as que aparecem sobre as teclas de 0 a 7.

Para fazer a tartaruga usar o lápis da cor certa, devemos digitar LAPISCOR e, em seguida, o número da tabela a seguir:

NÚMERO	COR
0	PRETO
1	AZUL
2	VERMELHO
3	LILÁS (MAGENTA)
4	VERDE
5	AZUL
6	AMARELO
7	BRANCO

Por exemplo:

```
?LIMPETELA
?COMLAPIS
?LAPISCOR 3
?DT 90 AV 50
?DT 90 AV 50
?DT 90 AV 50
?DT 90 AV 50
```

(O comando COMLAPIS é necessário apenas se você usou os comandos BORRACHA, INVELAPIS ou SEMLAPIS.)

Experimente variar a cor do lápis para ver o que acontece.

Rabisque bastante a tela. Experimente usar o comando INVELAPIS também.

Se você não lembra qual é a cor do lápis que está usando, pergunte à tartaruga. Basta digitar:

```
?ESCREVA COR?
```

O comando ESCREVA diz à tartaruga que coloque na tela alguma coisa. No próximo capítulo, você verá mais detalhes sobre essa ordem.

A palavra COR? não é um comando, mas uma operação. Por isso é necessário o uso do comando ESCREVA para que o resultado seja mostrado.

As operações são perguntas que você pode fazer à tartaruga.

Toda palavra, do vocabulário da tartaruga, seguida por uma interrogação, é uma operação.

Normalmente será necessário o uso do comando ESCREVA para que a tartaruga conte o que você quer saber.

Se você esquecer o comando ESCREVA (ou outro equivalente), a tartaruga responderá o seguinte:

NAO SEI O QUE FAZER COM 3

Pintando a Tela

A tela onde a tartaruga anda é branca. Mas você pode colorir o fundo onde aparecem os desenhos. Para isso, use o comando CORFUNDO, seguido do número da cor que você quer.

Experimente digitar:

```
?LIMPETELA
?COMLAPIS LAPISCOR 2
?CORFUNDO 6
?AV 80 EQ 90
?AV 80 EQ 90
?AV 80 EQ 90
?AV 80 EQ 90
```

Se tudo correu bem, você deve ter desenhado um quadrado vermelho sobre um fundo amarelo.

Experimente outras cores de lápis e de fundo. Tente fazer outros desenhos.

Se você esquecer o número da cor do fundo, pode perguntar à tartaruga:

?ESCREVA FUNDO?

(ou a abreviatura F?)

E a borda da tela? Vai continuar sempre branca? Não se preocupe, você pode alterar a cor dessa parte também. Experimente:

?CORBORDA 1

(Você pode usar a abreviatura, CB)

Viu o que aconteceu ?

Com tudo isso que aprendemos até agora, já podemos fazer uma série de coisas. Que tal ir ao próximo capítulo para ver como a tartaruga aprende a fazer coisas novas?

Comandos e operações vistos neste capítulo:

<i>COMANDOS</i>	<i>ABREVIATURA</i>	<i>EFEITO</i>
LIMPE	<i>Não há</i>	<i>Limpar a tela de desenhos sem mexer na tartaruga.</i>
ESCONDA	EC	<i>A tartaruga fica invisível, mas continua a desenhar.</i>
CENTRO	CT	<i>Colocar a tartaruga no centro da tela. Se o lápis estiver sendo usado, ela faz um traçados de seu trajeto.</i>
SEMLAPIS	SL	<i>A tartaruga anda sem traçar.</i>
COMLAPIS	CL	<i>A tartaruga volta a fazer traços.</i>
INVELAPIS	IL	<i>Desenhar onde não houver traços e apagar onde houver.</i>
CORFUNDO		<i>Pintar o fundo da tela.</i>
CORBORDA	CB	<i>Pintar a borda da tela.</i>
ESCREVA	ES	<i>A tartaruga escreve na tela.</i>
LAPISCOR		<i>A tartaruga troca a cor do lápis</i>

<i>OPERAÇÕES</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
COR?	—	<i>Diz qual é a cor do lápis.</i>
CORFUNDO?	F?	<i>Diz qual a cor é do fundo.</i>

**ENSINANDO
A TARTARUGA**

4

4. ENSINANDO A TARTARUGA

A tartaruga que mora na tela de seu computador é bastante inteligente. Podemos ensinar a ela muitas coisas. Neste capítulo você verá como ensinar coisas novas a ela.

Procedimentos e Procedimentos Primitivos

Imagine que você acabou de comprar um brinquedo de montar, uma miniatura com diversas peças que deverão ser coladas.

Ao abrir a embalagem, você encontra um mapa com as peças e uma série de instruções do tipo: “ponha o pino A no furo B”.

Essas instruções são um “procedimento de montagem” de seu modelo.

Certamente você não vai conseguir que a tartaruga monte um aeromodelo ou faça um bolo de chocolate, mas você vai ensinar uma série de procedimentos para ela.

Nos capítulos anteriores, você fez a tartaruga executar uma série de coisas. Isso foi possível graças a algumas palavras em Logo que você aprendeu. Essas palavras que a tartaruga entende e executa, são chamadas de “procedimentos primitivos” ou simplesmente “primitivos”.

A maneira como a tartaruga aprende uma coisa é semelhante ao modo como aprendemos uma palavra nova num dicionário. Quando não sabemos algum termo, olhamos no dicionário.

As definições do dicionário estão escritas com várias palavras. Na maioria dos casos, as palavras já são conhecidas e, lendo a definição, podemos aprender a nova palavra. Por exemplo, suponha que você não conheça a palavra “quadrado”. No dicionário provavelmente você vai encontrar:

QUADRADO. *figura que possui os quatro lados e os quatro ângulos iguais.*

Se você não souber o que significa “ângulo”, vai ter que consultar novamente o dicionário:

ÂNGULO. *figura formada por duas retas que têm um ponto em comum.*

Provavelmente com estas duas definições você já sabe o que significa “quadrado”. Se existisse mais alguma palavra desconhecida, o dicionário seria consultado mais uma vez.

A tartaruga também aprende novas palavras dessa forma. As ordens que você deu a ela até agora são formadas por procedimentos primitivos.

As palavras novas, chamadas de procedimentos ou definições, são formadas por uma combinação adequada de procedimentos primitivos.

Vamos ensinar a ela o que é um quadrado. Nós desenhamos um quadrado no capítulo anterior, não foi? O que fizemos foi algo assim:

?AV 50 EQ 90

?AV 50 EQ 90

?AV 50 EQ 90

?AV 50 EQ 90

A partir de agora, sempre que não houver algum desenho anterior na tela, você deve considerar que a tartaruga está visível, no centro da tela e que o lápis está sendo usado. Você será informado se tiver de considerar de outro modo.

A tartaruga desenhou um quadrado, mas ela não sabe disso! Experimente:

?QUADRADO

Ela vai responder:

AINDA NAO APRENDI QUADRADO

Você terá que dizer a ela o que a palavra quadrado significa. Diga a ela para aprender a palavra quadrado:

?APRENDA QUADRADO

O ponto de interrogação da linha seguinte muda para >. Tudo o que você digitar após este sinal será considerado parte da definição do procedimento.

Digite agora os procedimentos primitivos que formam a definição de quadrado:

>AV 50 EQ 90

>AV 50 EQ 90

>AV 50 EQ 90

>AV 50 EQ 90

Após ter digitado todas as ordens do procedimento, digite:

>FIM

Quando você der entrada <ENTER> a esta última linha, a tartaruga responderá:

QUADRADO DEFINIDO

Experimente agora digitar:

?QUADRADO

A tartaruga fará um quadrado igual àquele que você lhe ensinou. Isto, se você fez tudo certinho.

Se o quadrado não saiu como o esperado, é porque ocorreu um “grilo”, um bichinho que faz tudo para atrapalhar você e a tartaruga.

A caça ao grilo será explicada mais adiante. Por enquanto, a única maneira de eliminá-lo, é apagar o procedimento, com grilo e tudo.

Como exemplo, vamos ensinar um quadrado errado à tartaruga. Depois vamos apagá-lo:

?APRENDA GRILADO

>AV 50 DT 90

>AV 50 DT 90

>AV 50 DT90

>AV 50 DT 90

>FIM

Note que numa das linhas foi introduzido um grilo, de propósito. Não digitamos um espaço entre o comando DT e o número 90. Se você pedir para a tartaruga fazer GRILADO, ela mostrará a mensagem:

AINDA NAO APRENDI DT90 EM GRILADO

Não parece um bom português, mas o que ela está querendo dizer é que, no procedimento GRILADO, existia uma palavra que ela não conhecia.

Por enquanto, mostraremos como apagar o procedimento GRILADO.

Para apagar qualquer procedimento errado ou que você não deseje mais, sdê à tartaruga a ordem APAGUE (ou a sua abreviatura AP), seguida pelo nome do procedimento.

Por exemplo:

?APAGUE “GRILADO

Note as aspas colocadas antes do nome do procedimento. Você deve digitá-las sempre. Para consegui-las, aperte SYMBOL e P.

Se você se esquecer das aspas, a tartaruga responderá o seguinte:

GRILADO NAO E' SAIDA PARA APAGUE

Se você agora pedir para a tartaruga fazer GRILADO, ela lhe mostrará:

AINDA NAO APRENDI GRILADO

Para corrigir GRILADO, devemos digitar todo o procedimento novamente, desde a linha com a palavra APRENDA até a linha com a palavra FIM.

Uma outra maneira de corrigir o procedimento errado é ensinar à tartaruga a palavra que ela não conhece, dentro do procedimento. No nosso exemplo, a palavra desconhecida é DT90. Se digitarmos:

?APRENDA DT90

>DT 90

>FIM

Se digitarmos GRILADO, a tartaruga fará o quadrado corretamente. Mas, esta maneira de corrigir erros pode causar problemas. A memória da tartaruga é pequena. A tartaruga só pode aprender poucas palavras novas.

Pode ocorrer um outro tipo de grilo. Este é mais difícil de ser encontrado. Não faz a tartaruga mostrar nenhuma mensagem. Ela apenas não se comporta como você quer. Por exemplo:

?APRENDA SUPERGRILO

>AV 50 DT 90

>AV 50 DT 30

>AV 50 DT 90

>AV 50 DT 90

>FIM

Aqui, em vez de escrever DT 90 numa das linhas, escremos DT 30. A tartaruga nada responderá, mas não fará um quadrado. Agora, só resta apagar o procedimento.

Caçando Grilos

Apagar um procedimento e escrevê-lo novamente pode funcionar, mas, e se o procedimento for muito grande? Pode ser bastante trabalhoso reescrevê-lo!

Por sorte, você não precisará fazer isso. O Logo possui um procedimento especial, chamado EDITOR, cuja função é permitir a correção dos procedimentos que você escrever.

“Editar” é mais do que corrigir. Significa alterar, incluir coisas novas, melhorar e ... corrigir um procedimento que você tenha criado.

Vamos corrigir o procedimento GRILADO? Em primeiro lugar, devemos digitar:

?EDITE GRILADO

(Se quiser, use ED para abreviar, no lugar de EDITE)

Quando você apertar a tecla ENTER, a tela mostrada, quer seja de texto ou de desenhos, será apagada e, em seu lugar, aparecerá o procedimento escrito.

Na parte inferior da tela, estará a mensagem:

EDITOR LOGO-MULTISOFT INFORMATICA

A letra L, no canto inferior direito, continuará visível.

Os comandos do procedimento são mostrados em letras maiúsculas, mesmo que você os tenha escrito com letras minúsculas.

O cursor estará piscando sobre a primeira letra do procedimento. Para movê-lo, você deve usar as teclas com flechas (CAPS 5, 6, 7 ou 8).

Se você não apagou, o nosso procedimento GRILADO aparecerá da seguinte maneira:

**APRENDA GRILADO
AV 50 DT 90
AV 50 DT90
AV 50 DT 90
AV 50 DT 90
FIM**

Aperte 2 vezes CAPS 6 (flecha <↓>). O cursor descerá para a segunda linha do procedimento.

AV 50 DT90

Digite agora CAPS 8 <→> até que o cursor fique sobre o número 9.

AV 50 DT90

Aperte uma vez a tecla espaço. Deste modo você introduziu o espaço que estava faltando.

Este era o único grilo do procedimento. Se existissem outros, eles deveriam ser eliminados.

Quando estamos corrigindo um procedimento por meio do programa EDITOR, dizemos que estamos no “modo de edição”.

Agora, já que corrigimos o procedimento, eliminando o grilo, precisamos voltar ao “modo normal” de operação. Para isso, você deve apertar CAPS e SYMBOL ao mesmo tempo <EXTENDED MODE>.

A letra L, no canto inferior direito, mudará para E.

Se a letra E não aparecer, aperte CAPS e SYMBOL <EXTENDED MODE> outra vez.

Assim que a letra E aparecer, aperte a tecla C. Você verá a seguinte mensagem:

GRILADO DEFINIDO

Se o procedimento estiver certo, assim que você usá-lo novamente, a tartaruga fará o que você quer.

Outros Comandos do EDITOR

O EDITOR permite outros comandos para caçar e eliminar grilos. A tabela a seguir mostra todos os comandos de edição que o Logo possui:

COMANDOS DE EDIÇÃO	
COMANDOS	EFEITO
CAPS 5 <←>	Mover o cursor para a esquerda.
CAPS 6 <↓>	Mover o cursor para baixo.
CAPS 7 <↑>	Mover o cursor para cima
CAPS 8 <→>	Mover o cursor para a direita.
CAPS 0 <DELETE>	Apaga a letra à direita do cursor.
MODO E CAPS 5 <←>	Mover o cursor para o começo da linha.
MODO E CAPS 6 <↓>	Mover o cursor para o fim da tela.
MODO E CAPS 7 <↑>	Mover o cursor até o começo da tela.
MODO E CAPS 8 <→>	Mover o cursor até o fim da linha.
MODO E B	Mover o cursor para o começo do texto.
MODO E E	Mover o cursor para o fim do texto.
MODO E P	Mover o cursor para a página anterior.
MODO E N	Mover o cursor para a página seguinte.

cont.

COMANDOS	EFEITO
MODO E Y	Apagar a linha onde se encontra o cursor.
MODO E R	Recolocar a linha apagada por MODO E Y, no local onde se encontra o cursor.
MODO E C	Sair do modo de edição.

MODO E significa que *CAPS* e *SYMBOL* foram abaixadas ao mesmo tempo *<EXTENDED MODE>*. A letra no canto inferior direito da tela é *E*.

Comandos vistos neste capítulo:

<i>COMANDOS</i>	<i>ABREVIATURA</i>	<i>EFEITO</i>
<i>APRENDA</i>	—	Fazer com que a tartaruga aprenda um novo procedimento.
<i>FIM</i>	—	Marcar o final do procedimento.
<i>EDITE</i>	<i>ED</i>	Permitir mudar ou colocar novos comandos e corrigir um procedimento.

VARIÁVEIS

E LISTAS

5

5. VARIÁVEIS E LISTAS

Vamos começar definindo um quadrado:

```
?APRENDA QUADRADO  
>AV 50 DT 90  
>AV 50 DT 90  
>AV 50 DT 90  
>AV 50 DT 90  
>FIM
```

Suponha que você queira agora desenhar um quadrado um pouco maior, com o lado igual a 80, e não 50.

Podemos definir novamente um quadrado, agora com o nome de QUADRADO1. Em vez de fazermos isso passo a passo, vamos simplificar o trabalho, usando o EDITOR. Digite:

```
?EDITE QUADRADO
```

Veremos então o procedimento quadrado na tela do EDITOR:

```
APRENDA QUADRADO  
AV 50 DT 90  
AV 50 DT 90  
AV 50 DT 90  
AV 50 DT 90  
FIM
```

Leve o cursor até o primeiro espaço depois da letra O da palavra QUADRADO:

```
APRENDA QUADRADO ☐
```

Digite o número 1.

```
APRENDA QUADRADO1
```

Apertando ao mesmo tempo CAPS e 0 <DELETE> e usando as flechas, mude todos os valores 50 para 80.

Quando terminar, aperte CAPS SYMBOL <EXTENDED MODE> e, a seguir, C. Não se esqueça de verificar se a letra no canto inferior direito mudou para E. Será apresentada a mensagem:


```
QUADRADO1 DEFINIDO
```


Vamos experimentar para ver se deu certo?

Digite:

?QUADRADO
?QUADRADO1

Se você digitou tudo certo, verá dois quadrados, um dentro do outro, na tela.



E se você quisesse agora um quadrado com lado igual a 40? Podemos repetir o processo para um QUADRADO2.

Um quadrado com lado igual a 70 pediria um outro procedimento, e assim por diante, até que a memória da tartaruga não agüentasse mais.

Você deve estar imaginando que isso não parece muito prático. Você tem razão.

Existe uma outra maneira para fazer isso. Podemos escrever um procedimento onde o lado possa ser de vários tamanhos. Para isso, precisamos saber usar variáveis.

Variáveis

Quando criamos o procedimento QUADRADO1, a partir do procedimento QUADRADO, mudamos o nome do procedimento, e modificamos todas as linhas que tinham o valor do tamanho do lado.

Veja o procedimento primitivo AVANCE (AV). Para que a tartaruga entenda-o, é necessário dizer a ela quantos passos deve avançar. Isso é feito colocando-se depois de um espaço o número de passos.

Seria bom que o nosso quadrado tivesse um comportamento igual. Nós podemos fazer isso se usarmos uma variável.

Uma variável é um elemento do procedimento que varia como quisermos.

Podemos entender variável como uma gaveta que contém alguns objetos. Esta gaveta pode ser marcada por uma etiqueta com um nome escrito.

Podemos pôr ou tirar objetos desta gaveta, sem que a etiqueta seja mudada. O que vai mudar é o que está dentro dela, os objetos.

Vamos pôr uma etiqueta na gaveta que possui o valor do lado? Daremos a ela o nome LADO. Devemos também achar um jeito de colocar o nome da gaveta no procedimento.

Para marcarmos uma “gaveta” de um procedimento, devemos escrever, depois do nome do procedimento, o nome da etiqueta. Mas antes de escrever este nome, devemos digitar um espaço seguido por dois pontos.

Por exemplo (antes de digitar o exemplo, apague o procedimento QUADRADO, digitando APAGUE “QUADRADO)

?APRENDA QUADRADO :LADO

Só isso não basta. Esta variável deverá ser usada dentro do procedimento, para que tenha alguma utilidade. No nosso exemplo, ela deve ser usada pelo comando AVANCE:

?APRENDA QUADRADO :LADO

>AV :LADO DT 90

>AV :LADO DT 90

>AV :LADO DT 90

>AV :LADO DT 90

>FIM

Note o seguinte:

- 1) Colocamos dois pontos antes do nome da variável, todas as vezes em que ela foi usada.*
- 2) Não se esqueça de deixar um espaço antes de escrever os dois pontos.*
- 3) Não existe espaço entre o nome da variável e os dois pontos.*
- 4) O nome da variável pode ser escrito tanto em letras minúsculas como maiúsculas.*

Vamos usar este novo procedimento?

Digite:

?QUADRADO

A tartaruga responderá:

FALTAM DADOS EM QUADRADO


Isso aconteceu porque agora **QUADRADO** precisa de um dado, da mesma forma que os comandos **AV** e **DT**.

Se você quiser um quadrado com 50 passos de lado, deverá dizer isso à tartaruga:

?QUADRADO 50

Da mesma forma que os comandos **AV**, **EQ**, **DT** e outros, o procedimento **QUADRADO** deve ter um dado separado de seu nome por um espaço.

Suponha agora que você quer desenhar um retângulo, e não um quadrado. O retângulo é uma figura geométrica que tem os quatro ângulos iguais, sendo seus lados paralelos iguais, dois a dois. Serão necessárias duas variáveis diferentes. Por exemplo, **ALTURA** e **LARGURA**.



Vamos usar o **EDITOR** para escrever este procedimento?

?EDITE RETANGULO

Depois de ter apertado **<ENTER>**, digite o procedimento, linha por linha, como você faria normalmente. Não se esqueça de apertar **<ENTER>** ao final de cada linha.

```
APRENDA RETANGULO :ALTURA :LARGURA
AV :ALTURA DT 90 AV :LARGURA DT 90
AV :ALTURA DT 90 AV :LARGURA DT 90
FIM
```

Para sair do modo de edição, aperte **CAPS SYMBOL <EXTENDED MODE>**. A letra **E** aparecerá no canto inferior direito da tela.

Digite, então, **C**. A tartaruga responderá:

RETANGULO DEFINIDO

Experimente agora digitar:

?RETANGULO 80 50

Se você se esquecer de um dos valores, a tartaruga responderá:

FALTAM DADOS EM RETANGULO

Você pode usar quantas variáveis quiser, desde que, no momento em que for usar o procedimento, não se esqueça de fornecer todos os dados.

O Comando ATRIBUA

Você também pode armazenar alguma coisa numa variável, sem que seja preciso criar um procedimento para isso. Existe no Logo o procedimento primitivo ATRIBUA.

Um nome de variável é apenas um nome, enquanto não estiver associado a uma “gaveta”. No Logo, podemos associar um nome de variável a uma “gaveta” ao mesmo tempo em que colocamos uma coisa dentro dela.

Colocar alguma coisa na gaveta e etiquetá-la é chamado “atribuir”. Para isso temos o comando ATRIBUA.

A maneira como devemos usar o comando ATRIBUA é a seguinte:

- 1) Escrevemos a palavra ATRIBUA.*
- 2) Damos um espaço.*
- 3) Digitamos aspas.*
- 4) Escrevemos o nome de uma variável.*
- 5) Damos outro espaço.*
- 6) Digitamos aspas novamente.*
- 7) Digitamos o seu conteúdo (o que vai dentro da “gaveta”).*
- 8) Apertamos <ENTER>.*

O conteúdo de uma variável pode ser formado por letras ou números. Se forem usados números, não será preciso digitar as aspas.

Vamos usar uma variável que contenha letras? Escreva o seguinte procedimento:

```
?APRENDA ESCRITA :NOMES  
> ESCRIVA :NOMES
```

>FIM

A variável *NOMES* conterà as letras que você acrescentar depois do nome do procedimento, quando for usá-lo. Como se trata de letras, é obrigatório o uso de aspas. Por exemplo:

?ESCRITA "ANA

A tartaruga responderá mostrando na tela a palavra ANA.

Listas

Experimente agora digitar:

?ESCRITA "ANA MARIA

A tartaruga mostrará a palavra ANA e a seguir mostrará a mensagem:

AINDA NAO APRENDI MARIA

Para ela, MARIA é outro procedimento, ainda não aprendido.

Experimente agora:

?ESCRITA [ANA MARIA]

A resposta será:

ANA MARIA

Experimente também digitar:

?ESCREVA "ANA MARIA
?ESCREVA [ANA MARIA]

O comando *ESCREVA* diz à tartaruga que escreva algo na tela. A abreviatura deste comando é *ES*.

O que aconteceu em cada um dos casos? Qual seria a diferença entre uma e outra forma de escrever ANA MARIA?

Quando você usou aspas, estava considerando apenas uma palavra. O primeiro espaço encontrado indicou à tartaruga que se tratava do fim de uma palavra. Assim, o que viesse depois do espaço seria outra coisa, um procedimento por exemplo.

No segundo caso, a tartaruga encontrou o sinal [, que se chama "colchete". As palavras que estiverem entre colchetes formam uma lista, um conjunto de

objetos.

Uma lista pode conter palavras, letras, números e até outras listas (podemos ter uma lista de listas). Cada membro de uma lista é chamado de “item” desta lista.

Mostramos a seguir alguns exemplos de listas:

[CACHORRO GATO URSO LEAO]

[HOMEM MULHER CRIANCA]

[10 30 1000 1000]

Na primeira e na segunda lista, os itens são apenas palavras. A terceira lista é formada por números.

Observe agora as seguintes listas:

[15 HOMENS]

[[BOLA BONECA] [PATIM BICICLETA]]

[COMIDA [ARROZ FEIJAO] SAL 25 345]

A primeira delas é formada por um número seguido de uma palavra. Podemos ter listas mistas como esta, em que figuram números e palavras.

A segunda lista é uma lista de listas, ou seja uma lista onde seus membros são outras listas.

Você pode estar achando estranho que exista uma lista de listas. Mas isso não seria estranho se você fosse Papai Noel. Todo ano ele recebe inúmeras cartas com listas de presentes que ele deve entregar. Para se organizar, ele cria um “listão geral” que contém todas as listas. Assim, ele tem uma lista de listas.

Podemos ter também uma lista mista, onde os elementos podem ser palavras, listas ou números, como a terceira lista do exemplo.

Existe ainda a lista vazia, que não contém nenhum item. Ela é representada apenas por dois colchetes [].

Qualquer uma destas listas poderá ser atribuída a uma variável, quer num procedimento ATRIBUA, quer num procedimento que admita uma variável.

Por exemplo:

```
?APRENDA PAPAINOEL :PRESENTES
>ESCREVA [PAPAI NOEL, LEMBRE-SE DESTES PRESENTES:]
>ESCREVA :PRESENTES
>FIM
```

Note a lista interna do procedimento. Quando você for usar este procedimento, experimente colocar uma lista como dado. Por exemplo:

```
?PAPAINOEL      [BOLA      CARRINHO      PIANO]
```

Quando você apertar >ENTER<, aparecerá a seguinte mensagem:

```
PAPAI NOEL, LEMBRE-SE DESTES PRESENTES:
BOLA CARRINHO PIANO
```

Experimente também com outras listas, inclusive listas de listas.

Mais tarde mostraremos como usar as listas e manipular os itens delas.

Resumo do que você aprendeu neste capítulo

COMANDOS	ABREVIATURA	EFEITO
ATRIBUA	não há	Atribui um valor a uma variável

Conceito de variável e lista

Variável - elemento de um procedimento que varia. Uma variável é escrita à direita do nome do procedimento. Antes do nome da variável deve ser digitado um espaço seguido por dois pontos.

Lista - uma lista é um conjunto de elementos entre colchetes. Os elementos podem ser valores, nomes ou listas.

Item - um elemento de uma lista.

ESCREVENDO

6

6. ESCRREVENDO

Você já sabe que pode escrever tanto na tela de letras quanto na tela de desenhos. Agora vamos aprender mais coisas, para usar quando estamos escrevendo.

Apagador de Letras

A tartaruga tem um apagador que apaga todas as letras escritas na tela. Para que ela use esse apagador, você deve usar o comando *LIMPETEXTO* ou sua abreviatura *LTX*.

Experimente digitar o seguinte:

```
?TELATEXTO  
?ESCREVA "APAGADOR
```

Agora digite o comando *LIMPETEXTO* e veja o que acontece.

Brincando com as letras

Vamos fazer com que todas as letras fiquem piscando na tela? Digite:

```
?PISCA ESCRIVA "PISCANDO
```

Após o comando *PISCA*, tudo o que for pedido para ser mostrado na linha que contiver este comando será mostrado piscando, ou seja alternando a cor do fundo com a figura. Na linha seguinte tudo voltará ao normal.

Se você quiser desligar o "pisca-pisca" na mesma linha de comando, você deve digitar o comando *NORMAL*. Tudo voltará ao normal. Por exemplo:

```
?PISCA ESCRIVA "PISCANDO NORMAL ES "NORMAL
```

Que tal colorir as letras e a parte da tela onde elas aparecem?

Experimente digitar o seguinte:

```
?CORTSC [2 4] ES [VERDE SOBRE VERMELHO]
```

Você usou o comando *CORTSC*, que quer dizer cor texto sobre o campo. Este comando deve ser seguido por uma lista com dois números. O primeiro número indica a cor do fundo (campo). O segundo número indica a cor das letras (texto).

Existe outra forma de dar realce a uma palavra: o comando *BRILHO*.

Com *BRILHO 1*, a mensagem a ser mostrada terá seu fundo com o dobro do brilho normal. Este comando é restrito a uma única linha. Para desligar o *BRILHO* na mesma linha de comando, use *BRILHO 0*. Exemplo:

?BRILHO 1 ES "FULGURANTE BRILHO 0 ES "FOSCO

E se a tartaruga pudesse escrever em qualquer parte da tela? Vamos fazer um teste?

?LEVE [25 12] DIGITE "A

A letra A é escrita na coluna 25 da linha 12 da tela.

Você usou o comando LEVE. Este comando deve ser seguido por uma lista com dois números. O primeiro número indica a coluna, e o segundo número indica a linha da tela para onde o cursor é levado.

O número da coluna poderá variar de 0 a 30. O número da linha pode variar de 0 a 21.

Experimente digitar:

?ES CURSOR

Agora vamos escrever letras sobre letras usando o comando SOBRE. Digite:

? LEVE [25 10] ES "A SOBRE 1 LEVE [25 10] ES "C

Onde coincidirem os traços, eles serão apagados e onde não existir coincidência, o traço será mostrado.

SOBRE 0 desliga a sobreposição de caracteres.

A tartaruga vai dizer em que linha e coluna o cursor está.

Você poderá usar os comandos vistos neste capítulo dentro de um procedimento. Assim, vai perceber que brincar com as letras e números é tão divertido quanto desenhar.

Comandos e operações vistos neste capítulo

<i>COMANDO</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
LIMPETEXTO	LTX	<i>apaga a tela de texto</i>
PISCA		<i>apresenta a mensagem piscando</i>
CORTSC		<i>indica a cor das letras e do fundo das mensagens a serem escritas.</i>
LEVE	L	<i>leva o cursor para a posição Indicada.</i>
BRILHO		<i>faz com que o fundo de uma mensagem fique brilhante</i>
SOBRE		<i>permite a sobre posição de caracteres</i>

<i>OPERAÇÃO</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
CURSOR		<i>fornece a posição do cursor.</i>

**REVENDO O QUE A TARTARUGA
APRENDEU E GUARDANDO EM FITA**

7

7. REVENDO O QUE A TARTARUGA APRENDEU E GUARDANDO EM FITA

A memória é uma coisa fantástica. Ela deve ter surpreendido você em muitas ocasiões. Às vezes a gente não consegue se lembrar de um nome. Momentos depois, este nome pode vir espontaneamente.

Diversas vezes você deve ter se esquecido de coisas importantes. E deve ter chegado à conclusão de que não se pode confiar na memória.

Isto é verdadeiro também para o seu trabalho com o Logo. Se você criar muitos procedimentos, certamente vai esquecer os nomes deles. Também não vai lembrar dos comandos usados para defini-los.

Felizmente, o Logo possui alguns procedimentos primitivos destinados a “refrescar” a sua memória. Vamos vê-los?

Lembrando

Talvez você tenha esquecido os nomes dos procedimentos que a tartaruga já conhece. Não se preocupe, você pode pedir-lhe que mostre tudo o que ela sabe. Digite:

?MOTT

*(MOTT é a abreviatura de **MO**stre **T**odos os **T**ítulos dos procedimentos)*

A tartaruga mostrará todos os nomes de procedimentos. que ela aprendeu. Por exemplo:

**APRENDA QUADRADO
APRENDA QUADRADO1
APRENDA ESCRITA**

...

*Se você não se lembra das variáveis que forem definidas pelo comando **ATRIBUA**, pode revê-las por meio de:*

?MOTVAR

*(MOTVAR é a abreviatura de **MO**stre **T**odas as **VAR**iáveis)*

Todas as variáveis são apresentadas com o último valor que você atribuiu a elas. Por exemplo:

ATRIBUA "LADO "10
ATRIBUA "TEXTO "ABCDEF
ATRIBUA "NOME "DECIO
...

*E se você esquecer da definição de um procedimento? Use o comando **MODEF** (**MO**stre **DEF**inição).*

Este comando deve ser escrito da seguinte maneira:

?MODEF "nome do procedimento

Por exemplo:

?MODEF "QUADRADO

*Quando você apertar <ENTER>, serão mostradas todas as definições de **QUADRADO**:*

APRENDA QUADRADO
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
FIM

Você pode pedir para ver mais de um procedimento. Basta colocar os nomes deles entre colchetes. Por exemplo:

?MOSTRE [QUADRADO QUADRADO1]

*Se você usar o comando **MOTP** (**MO**stre **Todos os P**rocedimentos), verá as definições de todos os procedimentos. Exemplo:*

?MOTP

APRENDA QUADRADO
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
FIM

APRENDA QUADRADO1
AV 80 DT 90
AV 80 DT 90
AV 80 DT 90
AV 80 DT 90
FIM

APRENDA TEXTO
ESCREVA [BOM DIA COMO VAI]
FIM

Se você quiser ver tudo de uma vez (procedimentos e variáveis), use o comando TODOS:

?TODOS

O resultado apresentado será como o que mostramos a seguir:

APRENDA QUADRADO
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
AV 50 DT 90
FIM

APRENDA QUADRADO1
AV 80 DT 90
AV 80 DT 90
AV 80 DT 90
AV 80 DT 90
FIM

APRENDA TEXTO
ESCREVA [BOM DIA COMO VAI]
FIM

ATRIBUA "LADO "10
ATRIBUA "TEXTO "ABCDEF
ATRIBUA "NOME "DECIO

Gravando

Embora a tartaruga seja bastante inteligente, tem a memória é muito fraca. Assim que você desliga seu computador, tchau tudo o que ela aprendeu! Toda vez que isto acontecer, você terá que ensinar tudo de novo a ela.

Felizmente existe um modo de evitar ter de repetir as lições passadas. Você pode usar o gravador para guardar os procedimentos que a tartaruga aprendeu.

Gravando um Procedimento em Fita Cassete

Para esta operação, você vai precisar de um gravador e de uma fita sem uso.

Primeiro, ligue a tomada MIC do gravador à entrada MIC do computador. Se seu computador possui apenas uma entrada para cassete, ligue o cabo na entrada CASSETE.

Deixe assim, por enquanto. Você agora vai escolher os procedimentos que quer gravar.

Suponha que você queira guardar apenas o procedimento QUADRADO.

Em primeiro lugar, é necessário que você escolha um nome de arquivo para aquilo que vai gravar.

Toda vez que você grava alguma coisa do Logo em fita, nós dizemos que você está criando um arquivo.

Você deve conhecer aqueles arquivos de escritório. A função deles é a de armazenar papéis com informações. Um arquivo em Logo tem a mesma função que um arquivo de escritório, mas usa uma fita magnética para guardar as informações.

O nome de arquivo permite que você ache as informações mais tarde.

Escolha um nome de arquivo para gravar o procedimento QUADRADO na fita. É bom escolher um nome que lembre o que o arquivo contém. Por exemplo, QUAD. A seguir, digite o comando que manda a tartaruga gravar:

?GRAVE "QUAD "QUADRADO

Repare nos seguintes detalhes:

- 1) O nome de arquivo e o nome do procedimento vêm depois de aspas (").*
- 2) O nome do arquivo vem antes do nome do procedimento*

Quando você apertar <ENTER>, aparecerá a seguinte mensagem:

LIGUE O GRAVADOR E DIGITE ENTER

Siga esta instrução.

Aparecerão barras coloridas na borda da tela, durante alguns segundos.

Quando a gravação terminar, aparecerá o ponto de interrogação no começo da linha seguinte, indicando que você pode continuar a usar o Logo.

Agora escreva na etiqueta da fita o nome do procedimento que você gravou. Isso vai ajudá-lo no momento da procura.

Para gravar mais de um procedimento, agrupe-os numa lista, colocando-os entre colchetes. As aspas não devem ser usadas antes da lista. Por exemplo:

?GRAVE "QUAD2 [QUADRADO QUADRADO1]

O comportamento é o mesmo. Os procedimentos entre colchetes serão gravados na seqüência em que estão na lista.

Você também pode, ainda, o conteúdo do EDITOR (o último procedimento editado). Use, para isso:

?GRAVET

(T significa "textos")

Você pode, ainda, gravar tudo o que estiver em seu espaço de trabalho (procedimentos e variáveis) num único arquivo. Para isso, use:

?GRAVETUDO "TUDO

Tudo o que você ensinou à tartaruga, bem como o último valor atribuído a cada variável, por meio do comando ATRIBUA, é gravado na fita.

Muitas vezes, podemos criar um desenho muito bonito, através de comandos individuais, sem formar um procedimento. E, podemos salvar nossa "obra de arte" em fita, por meio do comando GRAVETELA.

A maneira de gravar a imagem da tela é idêntica aos procedimentos de gravação anteriores. Por exemplo:

**?QUADRADO QUADRADO1
?GRAVETELA "TELA**

O desenho formado pelos procedimentos QUADRADO e QUADRADO1 será gravado em fita com o nome de arquivo TELA.

Lendo o que Foi Gravado

Somente gravar não basta. Devemos também ter um modo de recuperar aquilo que gravamos. Isso é feito pelo comando LEIA.

Para recuperar um arquivo qualquer gravado em fita, você deve fazer o seguinte:

1) Ligue o cabo do gravador na tomada EAR do gravador e na tomada EAR do computador. Se o seu computador possuir uma única tomada para gravador, ligue o cabo na entrada marcada CASSETE.

2) Digite:

?LEIA "nome do arquivo

(o nome do arquivo deve ser o mesmo do arquivo gravado)

3) Aperte <ENTER>.

4) Ligue o gravador, através da tecla PLAY.

Aparecerão barras coloridas na borda da tela, indicando a atividade de carga.

Depois de alguns segundos, aparecerá o nome de arquivo do primeiro arquivo da fita, seguido por três letras. Estas letras indicam o tipo do arquivo.

Os tipos de arquivo são os seguintes:

LOG - procedimentos Logo. Podem ou não ser seguidos por variáveis.

TEL - tela

TXT - texto. O Logo considera tudo o que foi escrito em seu EDITOR como um texto.

BIN - arquivo binário (dados e programas em linguagem de máquina, que explicaremos mais tarde).

O procedimento primitivo LEIA lê o nome de qualquer arquivo, mas só pode carregar os arquivos LOG. Adiante mostraremos como ler os outros tipos de arquivo.

A busca do arquivo desejado só pára quando ele for encontrado. Assim que isso ocorrer, cada um dos procedimentos será armazenado. É mostrado o

seu título, seguido pela palavra DEFINIDO. Se houver variáveis, cada uma delas será mostrada depois do comando ATRIBUA e juntamente com seu conteúdo. Por exemplo:

QUADRADO DEFINIDO
QUADRADO1 DEFINIDO

ATRIBUA "LADO "10
ATRIBUA "TEXTO "ABCDEF
ATRIBUA "NOME "DECIO

Para ler um texto gravado pelo comando GRAVET, use o comando LEIAT. Este comando colocará o conteúdo da fita no EDITOR, como se fosse o último procedimento chamado pelo EDITOR.

Quando a carga terminar, o sinal de interrogação reaparecerá. Digite ED. Se a carga foi correta, o procedimento para edição deverá ser o que estava gravado na fita.

Uma tela gravada em fita é recuperada pelo procedimento primitivo LEIATELA. Deve-se usar o mesmo nome de arquivo para recuperar um desenho gravado desta forma.

O processo é semelhante ao que descrevemos para os outros procedimentos de leitura. A imagem é montada linha por linha, de cima para baixo, recobrindo qualquer desenho ou texto que esteja na tela.

Recomendamos que você tome os seguintes cuidados:

- 1) Antes de gravar a tela, esconda a tartaruga. Se isso não for feito, uma imagem da tartaruga será gravada.*
- 2) Chame a tela de desenhos antes de carregar qualquer tela.*
- 3) Se estiver usando a tela de desenhos, assim que o nome de um arquivo aparecer na tela, aperte <ENTER>. ISSO É INDISPENSÁVEL, senão o processo de carga é interrompido. Para evitar esquecimentos, mantenha o dedo sobre a tecla <ENTER> até que a carga da tela desejada comece.*

Comandos vistos neste capítulo

<i>COMANDO</i>	<i>EFEITO</i>
MOTT	<i>Mostra todos os títulos dos procedimentos</i>
MOTVAR	<i>Mostra os nomes de variáveis e seu conteúdo</i>
MODET	<i>Mostra as definições dos procedimentos.</i>
MOTP	<i>Mostra todos os procedimentos e todas as variáveis com seus conteúdos.</i>
TODOS	<i>Mostra todos os procedimentos e todas as variáveis.</i>
GRAVE	<i>Grava um ou mais procedimento.</i>
GRAVET	<i>Grava o texto do EDITOR.</i>
GRAVETELA	<i>Grava uma tela.</i>
LEIA	<i>Lê um arquivo com procedimento ou variáveis</i>
LEIAT	<i>Lê um texto e o armazena no editor.</i>
LEIATELA	<i>Dê uma tela e a coloca no vídeo.</i>

ESTRUTURANDO

8

8. ESTRUTURANDO

A esta altura você já deve estar fazendo várias coisas em conjunto com a sua tartaruga.

Como você deve ter ensinado muitas coisas para a sua tartaruguinha, deve estar começando a sentir alguns problemas. Por exemplo, deve estar achando cansativo escrever o procedimento QUADRADO, repetindo quatro vezes a mesma linha. E se não fosse uma figura com quatro lados, mas com 20? Digitar vinte vezes a mesma coisa não é muito agradável.

A tartaruga aceita que você mande-a repetir alguma coisa. O comando usado para isto é REPITA. Para usá-lo, você deve fornecer o número de vezes e o que deve ser repetido como dado para a tartaruga.


Os procedimentos que serão repetidos deverão estar organizados numa lista. Experimente:

```
?APRENDA QUAD :LADO  
>REPITA 4 [AV :LADO DT 90]  
>FIM
```

Bem mais fácil, não?

Que tal outro exemplo? Vamos desenhar uma figura com doze lados iguais:


```
?APRENDA DOZELADOS :LADO  
>REPITA 12 [AV :LADO DT 30]  
>FIM
```



Vamos fazer uma outra experiência. Veja o que acontece:

```
?APRENDA CRISTAL :LADO  
>REPITA 24 [DOZELADOS :LADO EQ 15]  
>FIM
```

Experimente com os seguintes valores: 20, 30, 10



Agora vamos usar o comando REPITA com um texto. Vamos escrever uma palavra vinte vezes em seguida na tela:

```
?APRENDA ESCRITA :PALAVRAS  
>REPITA 20 [ESCREVA :PALAVRAS]  
>FIM
```

Teste este procedimento com seu nome e com uma frase (não se esqueça das aspas, quando for uma única palavra, ou dos colchetes, quando for uma frase).

Você deve ter percebido que o comando REPITA facilita muito seu trabalho de digitação e, além disso, é um recurso bastante poderoso para escrever novos procedimentos.

Existe um outro recurso bastante interessante, chamado “recursividade”. Para você entendê-lo, vamos contar-lhe uma pequena história:

O Gênio da Garrafa

Um pobre pescador chamado Kalil passeava numa praia deserta da Arábia. Caminhava pela beira-mar, pensando em seus problemas e nas formas de solucioná-los. De repente avistou um objeto boiando na água. Apanhou-o. Era uma garrafa, muito bem arrolhada.

Curioso, Kalil tentou ver o seu conteúdo. A garrafa era muito antiga e estava muito suja para que ele pudesse ver por fora o que ela continha.

A solução foi abrir a garrafa.

Qual não foi sua surpresa, quando uma fumaça espessa saiu e, diante de seus olhos, um enorme gênio apareceu!

O gênio agarrou-o, levou-o até perto do rosto e falou-lhe:

— Estou preso aqui faz mil anos. Você me salvou. Agora vou lhe conceder três desejos. Mas, ao término do terceiro pedido, vou matá-lo por ter demorado tanto a aparecer!

Kalil tinha ouvido muitas histórias sobre gênios e sabia que eles cumpriam o que prometiam, tanto para o bem, como para o mal.

Ele pensou, pensou, e formulou os seguintes pedidos:

— Eu quero que me ponha no chão.

O gênio obedeceu e disse:

— Você fez seu primeiro pedido.

Kalil, então, pediu:

— Dê-me sua palavra de que você realizará meu terceiro desejo do modo que eu indicar.

— Você fez o segundo pedido. Assim será feito.

No terceiro pedido, Kalil fez uma pausa, pensou um pouco e pediu:

— Quero que você me conceda mais três pedidos.

O gênio ficou bastante zangado, mas respondeu:

— Concedido!

E assim, Kalil fez mais três pedidos. No terceiro, pediu novamente mais três pedidos. E assim, sucessivamente, transformando o gênio em seu escravo por toda a sua vida.

A Recursividade

Esta pequena história ilustra um dos recursos mais importantes do Logo, a recursividade.

Kalil usou a recursividade, pedindo sempre mais três desejos, quando fazia o

terceiro pedido. No Logo, você pode fazer um procedimento chamar-se a si próprio e, desta forma, utilizar a recursividade.

Experimente:

```
?APRENDA SIPROPRIO
>ESCREVA [EU CHAMO A MIM MESMO]
>SIPROPRIO
>FIM
```

NÃO execute ainda!

Este procedimento, em sua terceira linha, chama a si próprio. Se você pedir à tartaruga que o execute, ela o fará por muito tempo. Felizmente existe um meio de pará-la: pressione CAPS e SPACE ao mesmo tempo <BREAK>.

A tartaruga interrompe o que estava fazendo e diz o seguinte:

PAREI!!!

<BREAK> permite que você pare qualquer procedimento. Se for usado no modo de edição, interromperá o processo, fazendo a tartaruga esquecer todas as correções.

Digite agora:

```
?SIPROPRIO
```

Na tela você verá:

```
EU CHAMO A MIM MESMO
```

Interrompa o procedimento com <BREACK>.

Experimente também:


```
?APRENDA RECURSIVO
>AV 5 EQ 1
>REC
>FIM
```

A Espiral Quadrada

Vamos fazer uma espiral quadrada? Digite:

?APRENDA ESPIRAL

```
> AV 10 DT 90
> AV 15 DT 90
> AV 20 DT 90
> AV 25 DT 90
> AV 30 DT 90
> AV 35 DT 90
> AV 40 DT 90
> AV 45 DT 90
> AV 55 DT 90
> FIM
```



Este não parece um meio muito racional de construir uma espiral. Será que não existiria outra maneira de fazer isso sem termos de digitar tanto?

Existe sim! Preste um pouco de atenção ao procedimento ESPIRAL.

Repare que cada linha é igual à anterior, sendo diferente apenas num detalhe. O número de passos da tartaruga no procedimento AV é igual ao da linha anterior mais cinco. Poderíamos contar a ela que isto está acontecendo.

Para isso, lançaremos mão de um procedimento recursivo e de uma operação aritmética.

Veja:

```
?APRENDA ESPIRAL :A
> AV :A DT 90
> ESPIRAL :A + 5
> FIM
```

Digite agora:

```
?ESPIRAL 10
```

Quando quiser parar, use <BREAK>.

Se você não parar o procedimento, a memória da tartaruga vai ficar lotada. Então ela mesma parará o procedimento e mostrará a seguinte mensagem:

SEM MEMORIA DISPONIVEL

Veja a linha anterior à palavra FIM. Ela foi usada para dizer à tartaruga que esta devia aumentar para mais cinco o número de passos a serem dados.

Nós usamos para isso a operação aritmética da adição, representada pelo sinal +.

No próximo capítulo veremos mais detalhes sobre operações aritméticas, por enquanto usaremos apenas a operação de soma.

Vamos fazer outra espiral?

```
?APRENDA ESPIRAL2 :A :ANGULO  
>AD :A EQ :ANGULO  
>ESPIRAL :A + 1 :ANGULO  
>FIM
```

Experimente com:

```
?ESPIRAL2 1 30
```

```
?ESPIRAL2 1 45
```

```
?ESPIRAL2 1 60
```

```
?ESPIRAL2 1 90
```

Não se esqueça de limpar a tela quando terminar cada experiência.

Você agora deve estar pensando numa maneira de parar seus procedimentos, sem ter que usar <BREAK>.

Existe uma maneira de fazer isso. Veja no próximo capítulo.

Resumo de que foi visto neste capítulo

<i>COMANDOS</i>	<i>EFEITO</i>
REPITA	<i>Permite que uma lista de comandos seja repetida várias vezes.</i>

<i>TECLA</i>	<i>EFEITO</i>
CAPS SPACE <BREAK>	<i>Interrompe a execução ou a edição de um procedimento.</i>

Conceitos

Recursividade - A propriedade de recursividade é permitir que um procedimento chame a si mesmo, executando-se infinitas vezes.

OPERAÇÕES ARITIMÉTICAS

E COMPARAÇÕES

9

9. OPERAÇÕES ARITMÉTICAS E COMPARAÇÕES

A tartaruga também conhece aritmética!

Ela é capaz de fazer as seguintes operações:

operação	símbolo
divisão	/
multiplicação	*
subtração	-
soma	+

Os símbolos são colocados entre os dois números. Estes números são chamados “operandos”.

O sinal menos (–) precisa ser escrito, seguidos por um espaço antes do número.

Experimente:

?ESCREVA $3 + 5$
8

?ES $4 * 25$
100

?ES $45 < 10$
35

(não esqueça do espaço entre o número 10 e o sinal menor).

?ES $25 / 5$
5

Você pode usar mais de dois operandos, fazendo uma “expressão”:

?ES $5 * 2 + 3 - 5$
18

A tartaruga resolve as expressões numa certa ordem. Primeiro ela fará as divisões, a seguir as multiplicações, depois as subtrações e, por último, as somas.

Esta ordem pode ser modificada, se você colocar a expressão que deve ser calculada entre parênteses. Por exemplo:

?ES (5 - 2 + 3) * 5
30

Você **NÃO** pode usar mais de um par de parênteses dentro de outro par de parenteses numa mesma expressão. Por exemplo:

?ES ((10 - 2) + 10) * 10

A tartaruga responderá:

MUITOS PARENTESSES

As operações de soma, subtração e multiplicação podem ser escritas de outra maneira. Para isso usamos uma abreviatura do nome da operação, o prefixo.

Existem os seguintes prefixos:

SOME	soma
MULT	multiplicação
DIV	divisão

Devemos escrevê-los da seguinte maneira:

?ES prefixo valor1 valor2

Por exemplo:

?ES MULT 3 5
15

A tartaruga fará a multiplicação de 3 por 5 e escreverá o resultado na tela.

Qualquer uma das operações pode ser feita com números fracionários positivos ou negativos. Mas, em vez da vírgula, você deve usar o ponto para separar a parte decimal da parte inteira. Por exemplo:

?ES MULT 3.5 6.2
21.7

?ES 3.2 + 10
13.2

?ES 10/3
3.33333333

Você também pode usar variáveis numa expressão. Por exemplo:

```
?ATRIBUA "A 2
?ATRIBUA "B 4
?ATRIBUA "C 5

?ES :A + :B * :C
22
```

Uma variável pode conter o resultado de uma expressão:

```
?ATRIBUA "A 3+2
?ES :A
5
```

Ou ainda pode conter o resultado de uma expressão com diversas variáveis, inclusive ela mesma (desde que tenha sido definida antes). Por exemplo:

```
?ATRIBUA "A 2
?ATRIBUA "B 2
?ATRIBUA "C 10

?ATRIBUA "A :A + :B * 105 / :C
?ES :A
23
```

Usando um Pouco de Lógica

“Isso não é lógico!” Dizemos isso quando alguém fala alguma coisa sem sentido ou fora da realidade. Geralmente quando dizemos isso a alguém, explicamos porque determinada afirmação não é lógica. Isso se chama “argumentar”. As explicações que damos são chamadas de “argumentos”.

A Lógica é a ciência que verifica a validade de um argumento, em relação à verdade ou falsidade de suas afirmações.

O Logo verifica se uma determinada afirmação é verdadeira ou falsa e apresenta a mensagem CORRETO, se a afirmação é verdadeira, ou FALSO, em caso contrário.

Vamos verificar? Digite o seguinte procedimento:

```
?APRENDA TABUADA :A :B :C
>ES :A * :B = :C
>FIM
```


Este procedimento serve para “tomar-lhe a tabuada”. Para usá-lo você deve digitar o nome do procedimento seguido por três valores. O primeiro e o segundo valores são os números que serão multiplicados, e o terceiro é o valor que você acha que é o certo. Por exemplo:

?TABUADA 3 7 21

A tartaruga responderá CORRETO, já que o resultado está certo. Realmente, 3 vezes 7 é igual a 21.

Experimente agora:

?TABUADA 3 4 20

A resposta será FALSO, já que o resultado está errado: 20 não é igual a 3 vezes 4.

A tartaruga faz uma comparação entre o resultado da operação e o valor que você indicou como correto. Se eles forem iguais, a tartaruga responderá com a palavra CORRETO. Se não forem, a resposta será FALSO.

A tartaruga pode fazer mais duas operações de comparação com números. Ela pode verificar se um número é maior que outro ou se um número é menor que outro. Por exemplo:

?ES 10 > 5
CORRETO

?ES 3 < 8
CORRETO

?ES 2 < 1
FALSO

?ES 2 > 10
FALSO

Podemos comparar dois números, duas variáveis ou o resultado de duas expressões. Por exemplo:

?ES 3 + 4 > 5 + 1
CORRETO

?APRENDA MENOR :A :B
>ES :A < :B
>FIM

Usando Lógica para Parar um Procedimento

Dissemos, no capítulo anterior, que um procedimento recursivo *funciona até* que a memória da tartaruga esgote-se. Você só pode parar um procedimento desse tipo através de `<BREAK>`.

Nós dissemos também que existe uma outra maneira de fazer um procedimento recursivo parar. Essa maneira é usar comparações.

Você viu há pouco que a tartaruga verifica se uma comparação está correta ou não. Podemos usar esta capacidade da tartaruga, para que ela verifique quando um procedimento deve parar. Para isso, usaremos um procedimento primitivo: `SE`.

`SE` é uma palavra que indica uma condição a ser satisfeita. Veja a frase abaixo:

Se chover, usaremos guarda-chuva.

Ocorrendo a condição indicada por `Se` (chover), a ação de usar um guarda-chuva será realizada.

Da mesma forma, o procedimento `SE` faz com que, ocorrendo uma condição, uma ação seja realizada. Por exemplo:

```
?APRENDA VERIFIQUE :A :B
>SE :A > :B [ES :A [e' maior que ] ES :B]
>FIM
```

Neste procedimento, se a variável `:A` for maior que a variável `:B`, uma mensagem será escrita na tela.

Poderíamos escrever o procedimento para verificar se `:A` é menor que `:B`. Era só ter escrito:

```
?APRENDA VERIFIQUE2 :A :B
>SE :A < :B [ES :A [e' menor que ] ES :B]
>FIM
```

Poderíamos ainda ver se ambos eram iguais. Vamos fazer um procedimento completo que verifique tudo?

```
?APRENDA VERIFIQUE3 :A :B
>SE :A > :B [ES :A ES [e' maior que ] ES :B]
>SE :A < :B [ES :A ES [e' menor que ] ES :B]
>SE :A = :B [ES :A ES [e' igual a ] ES :B]
>FIM
```

Como poderíamos lançar mão desta capacidade da tartaruga para interromper um procedimento recursivo? Lembra-se do procedimento ESPIRAL?

```
?APRENDA ESPIRAL :A  
>AV :A DT 90  
>ESPIRAL :A + 5  
>FIM
```

Para interromper este procedimento, podemos acrescentar, depois da linha que diz à tartaruga que vá à frente, a seguinte linha:

```
>SE :A > 100 [PARE]
```

O procedimento ficaria assim:

```
?APRENDA ESPIRAL :A  
>AV :A DT 90  
>SE :A > 100 [PARE]  
>ESPIRAL :A + 5  
>FIM
```

O comando PARE indica à tartaruga que pare o que estiver fazendo. Assim que a variável :A chegar ao valor 101, o procedimento não chamará a si próprio outra vez.

Você poderia colocar a linha condicional antes da linha que contém a primeira ação. O procedimento ficaria assim:

```
?APRENDA ESPIRAL :A  
>SE :A > 100 [PARE]  
>AV :A DT 90  
>ESPIRAL :A + 5  
>FIM
```

No primeiro procedimento, a ação AV :A DT 90 é executada uma vez, mesmo que o dado introduzido seja maior que 100. No segundo procedimento, se o dado introduzido for maior que 100, nenhuma ação será realizada.

Recomendamos o uso deste recurso para interromper seus procedimentos recursivos, a menos que você queira que a tartaruga faça a mesma coisa até ficar com a memória lotada.

Expressões Lógicas

Verifique agora a seguinte sentença:

“Se amanhã chover e eu tiver que sair de casa, usarei guarda-chuva”.

De acordo com esta frase, você usará guarda-chuva se chover amanhã E se você tiver que sair. Apenas se ocorrerem (se são verdadeiras) as duas coisas juntas é que você usará o guarda-chuva. Veja esta outra frase:

“Se eu for ao cinema ou ao teatro, voltarei tarde para casa”.

Basta que ocorra (que seja verdadeira) uma das duas situações (ir ao cinema OU ir ao teatro) para que você volte tarde.

A diferença entre os dois tipos de frase é o elemento de ligação entre as duas ações que devem ocorrer para que a terceira aconteça. No primeiro tipo, o elemento de ligação é a palavra “E”. No segundo, é a palavra “OU”.

Quando usamos E, apenas se as duas ações ligadas por ele forem verdadeiras é que a ação principal ocorrerá.

Quando usamos OU, basta que uma das ações seja verdadeira, para que a ação principal ocorra.

Observe agora:

“Se NÃO chover, sairei.”

A palavra NÃO modifica o valor de verdade de uma frase. No exemplo, se chover, a ação de sair não acontecerá.

Estas três palavras (OU, NÃO e E) são usadas no Logo também e são conhecidas como “operadores lógicos”, ou seja, funcionam como as operações aritméticas.

Por meio destes operadores, a tartaruga verifica o valor de verdade de duas comparações.

A operação E só dá como resultado a palavra CORRETO, se as duas comparações forem verdadeiras. Por exmplo:

?ES E 2>3 2=2
FALSO

?ES E 4<5 6>2
CORRETO

?ES E 3<5 5>6
FALSO

?ES E 1=1 100>1
CORRETO

A operação OU fornece como resultado a palavra CORRETO, se pelo menos uma das comparações for verdadeira. Por exemplo:

?ES OU 10>30 1=1
CORRETO

?ES OU 2<5 7>1
CORRETO

?ES OU 3>5 5>6
FALSO

?ES OU 1=10 100<10
FALSO

A operação NAO (não se usa o til no Logo) inverte o valor lógico de uma comparação. Por exemplo:

?ES NAO 1=1
FALSO

?ES NAO 2=0
CORRETO

Resumo dos conceitos vistos neste capítulo

OPERAÇÕES	SIMBOLO	SIGNIFICADO
Divisão	/	Realiza a divisão entre dois valores numa expressão.
Multiplicação	*	Realiza a multiplicação entre dois valores numa expressão.
Subtração	–	Realiza a subtração entre dois valores numa expressão.
Soma	+	Realiza a soma entre dois valores numa expressão.
SOME		Realiza a soma de dois valores, mas não pode ser usado em uma expressão
DIV		Realiza a divisão entre dois valores, mas não pode ser usado em uma expressão.
MULT		Realiza a multiplicação entre dois valores, mas não pode ser usado em uma expressão.

COMANDOS	SIGNIFICADO
E	Realiza a função lógica E entre dois valores lógicos.
OU	Realiza a função lógica E entre dois valores lógicos
NAO	Altera o valor lógico de correto para falso e vice-versa de uma expressão lógica
SE	Executa uma lista de comandos se uma determinada condição for correta.

DESENHANDO UMA CASA
COM JARDIM

10

10. DESENHANDO UMA CASA COM JARDIM

Os procedimentos que vimos até aqui são simples, sem muitos comandos. Agora faremos um procedimento bem complicado. Vamos criar uma casa com jardim.

Existem duas maneiras para fazer isso. A primeira delas é tentar escrever o procedimento de uma vez só. A outra é dividi-lo em partes pequenas, elaborar e testar cada uma delas e, depois, construir um procedimento que reúna tudo.

As duas maneiras são equivalentes... se o procedimento der certo da primeira vez.

No primeiro método, é difícil encontrar um grilo. No segundo, a coisa é mais fácil.

Um procedimento organizado em partes pode ser corrigido. Basta que localizemos o grilo e que o eliminemos. Num procedimento grande, sem repartições, dá tanto trabalho achar grilo, que às vezes é preferível escrever tudo outra vez!

Dividindo o Terreno

Podemos começar escrevendo num papel o que iremos desenhar na tela. Por exemplo:

- 1) uma casa*
- 2) duas árvores*
- 3) uma cerca*
- 4) um sol*

Podemos escrever um procedimento para cada uma destas partes e um procedimento maior para reunir as quatro partes. Vamos começar pela casa.

A casa, por sua vez, pode ser dividida em outras partes:

- 1) Um quadrado, representando a fachada*
- 2) Um quadrado menor, representando a janela*
- 3) Um retângulo, representando a porta*
- 4) Um triângulo, representando o telhado*

Não precisamos criar um procedimento para o quadrado maior, outro para o quadrado menor e mais outro para o retângulo. Temos apenas que criar o procedimento retângulo, com duas variáveis. Quando formos usar os quadrados, basta fazer a altura igual à largura.

Escreva o seu procedimento RETANGULO:

```
?APRENDA RETANGULO :A :L  
>REPITA 2 [AV :A DT 90 AV :L DT 90]  
>FIM
```

Precisamos agora definir o procedimento TRIANGULO. Um triângulo é uma figura de três lados e três ângulos. Se o triângulo tem os três lados e os três ângulos iguais, ele é chamado equilátero. Como é mais fácil desenhar um triângulo equilátero, é o que faremos.

No quadrado usamos como ângulo 90 graus. Que ângulo usaremos? Analise o procedimento QUADRADO, usado por nós várias vezes. A tartaruga anda um trecho; vira 90 graus; anda outro trecho; vira mais 90 graus; anda mais um trecho; vira 90 graus; e finalmente anda mais um trecho e vira 90 graus. Ao todo, ela gira 360 graus.

$$\text{Giro} = 90 + 90 + 90 + 90 = 360 \text{ graus}$$

Ela gira quatro vezes, e a soma dos ângulos é igual a 360 graus.

$$\hat{\text{Ângulo}} = 360/4$$

Como o triângulo possui três lados, o ângulo do giro da tartaruga deverá ser:


$$\hat{\text{Ângulo}} = 360/3 = 120 \text{ graus}$$

Vamos então definir o triângulo:

```
?APRENDA TRIANGULO :C  
>REPITA 3 [AV :C DT 120]  
>FIM
```

Experimente com:

```
?TRIANGULO 45
```



Podemos evitar que o triângulo fique inclinado, se mandarmos a tartaruga virar 30 graus:


?LT DT 30 TRIANGULO 45

Vamos montar uma parte da casa?

*?APRENDA CASA
> RETANGULO 45 45
> DT 30
> TRIANGULO 45
> FIM*

Será que está certo? Experimente:

?CASA



O telhado saiu embaixo, mas não era isso que queríamos. Experimente agora:

*?ED CASA
APRENDA CASA
RETANGULO 45 45
AV 45
DT 30
TRIANGULO 45
FIM*

Quando acrescentamos a linha AV 45, levamos o telhado para cima da casa.

Precisamos agora acrescentar a porta e a janela. Devemos recolher o lápis e levar a tartaruga até a posição onde queremos a porta; colocar o lápis; desenhar a porta e fazer o mesmo com a janela.

Nosso procedimento CASA ficará assim:

*?ED CASA
APRENDA CASA
RETANGULO 45 45
AV 45*

DT 30
TRIANGULO 45
SEMLAPIS CENTRO
DT 90 AV 10 EQ 90
COMLAPIS RETANGULO 20 10
SL AV 40 DT 90 AV 15
CL RETANGULO 15 15
FIM

O comando CENTRO leva a tartaruga ao centro da tela, na posição que ela ocupava no início.

Uma parte do projeto já está pronta. Apague a tela e prepare o procedimento ARVORE.

Nossa sugestão é:

?APRENDA ARVORE
> AV 50
> REPITA 36 [AV 30 VT 30 DT 10]
> FIM

Podemos agora contruir uma cerca:

?APRENDA CERCA
> REPITA 45 [AV 10 VT 10 DT 90 AV 5 EQ 90]
> FIM

Por fim, vamos desenhar o sol:


?APRENDA SOL
> REPITA 26 [AV 15 VT 15 DT 18]
> FIM

Vamos agora construir um procedimento que use cada um destes subprocedimentos, colocando as coisas no seu devido lugar. Não nos esqueceremos das cores também:

?APRENDA JARDIM
> ESCONDA
> CL LAPISCOR 2
> CASA
> SL CENTRO EQ 90 AV 35 DT 90
> CL LAPISCOR 4
> ARVORE
> SL CENTRO EQ 90 AV 65 DT 90 VT 20
> CL ARVORE

> SL CENTRO AV 70 EQ 90 AV 100
> CL LAPISCOR 6
> SOL
> SL CENTRO VT 40
> CL LAPISCOR 0
> CERCA
> ES [BEM VINDO AO MEU JARDIM]
> FIM

O procedimento está pronto. Agora é só usá-lo.



O ESPAÇO DA TARTARUGA


11

11. O ESPAÇO DA TARTARUGA

Em muitos procedimentos você tem usado os comandos *DIREITA* e *ESQUERDA*.

Os valores que você fornece à tartaruga fazem-na girar e apontar para uma determinada direção.

Na tela de desenhos, quando chamamos a tartaruga pela primeira vez ou quando a limpamos da tela, ela surge no centro, apontando para cima. Podemos considerar que ela está apontando para o norte. Se dermos a ela o comando *DIREITA* 90, a tartaruga passará a apontar para o leste. Se mandarmos ela girar mais 90 graus à direita, ela apontará para o sul; mais 90 graus, e ela apontará para o oeste; e, para finalizar, com mais 90 graus, ela apontará novamente para o norte.




Se somarmos todos os giros à direita, o resultado será 360 graus. Qualquer valor de giro à direita maior que 360 equivale ao valor dos graus menos 360. Por exemplo, *DT 450* é o mesmo que digitar *DT 90*, pois:

$$450 - 360 = 90$$

Todas as medidas de giro são dadas em graus. Grau é unidade de medida do giro.

Lembra-se de como construímos a árvores e o sol do nosso jardim? Nós fizemos a tartaruga ir um pouco à frente; dar um giro à direita; andar um tanto à frente e repetir o processo. Experimente:

?CENTRO ESCONDA AV 80 VT 80 DT 30 AV 80 VT 80 CENTRO



A tartaruga traçará uma reta; voltará ao ponto de origem e traçará outra reta, após um giro de 30 graus. Estas duas retas formam o que chamamos, em geometria, de “ângulo”.

A medida do giro que a tartaruga tem de dar para ir da primeira reta à segunda é idêntica à medida do ângulo formado por estas retas.

Podemos perguntar à tartaruga qual o ângulo do giro que ela deu. Para isso usamos a operação *GIRO?*.

GIRO? fornecerá quantos graus a tartaruga girou à direita, em relação à posição central (mesmo que ela tenha girado à esquerda). Experimente:

```
?CENTRO DT 30
?ESCREVA GIRO?
30
```

```
?CENTRO EQ 90
?ESCREVA GIRO?
270
```

```
?CENTRO DT 360
?ESCREVA GIRO?
0
```


Existe um outro meio de fazer a tartaruga virar à direita: o comando *GIRE*. *GIRE 90* é equivalente a *DIREITA 90*. Você pode usar tanto um como o outro, indiferentemente.

A Posição da Tartaruga

Além de girar, a tartaruga anda pela tela. Se não soubermos onde ela está (por exemplo, quando ela está escondida), podemos perguntar por ela.


No Logo a posição da tartaruga é fornecida por meio de coordenadas cartesianas.

Podemos imaginar a tela por onde a tartaruga movimenta-se como tendo dois eixos passando pelo centro: um eixo vertical e outro horizontal.



O eixo horizontal normalmente é chamado de eixo X. E o eixo vertical é chamado de eixo Y.

A tela da tartaruga pode ser representada por uma rede, dividida em pequenos quadradinhos, como um papel quadriculado. Cada quadradinho possui dois valores chamados de coordenadas, os valores X e Y.



Quando a tartaruga está no centro, dizemos que está na origem e que possui as coordenadas $X = 0$ e $Y = 0$.

A coordenada X indica quantos passos a tartaruga tem de andar na horizontal, a partir da posição central, até atingir um certo ponto horizontal.

A coordenada Y indica quantos passos a tartaruga tem de andar na vertical, a partir da posição central, até atingir um certo ponto vertical.

Você pode perguntar à tartaruga qual a posição onde ela está. A operação POS fornece os valores de X e Y, nesta ordem. Por exemplo:

```
?AV 50 DT 90 AV 30
?ES POS
50 30
```

Você pode também perguntar a posição de cada uma das coordenadas em separado:

```
?ES X?
50
```

```
?ES Y?
30
```

Se a tartaruga estiver no lado oeste da tela, o valor de X é negativo. Da mesma forma, se a tartaruga estiver na parte sul da tela, o valor de Y é negativo.

Você pode colocar a tartaruga em qualquer ponto da tela, se disser a ela o valor das coordenadas da posição desejada. Ela irá para a posição onde estiver

até o local indicado, fazendo um traçado de seu caminho. O comando para isso é POSXY.

POSXY Coloca a tartaruga num ponto de coordenadas X Y, traçando o trajeto dela.

Os valores de X e Y deverão estar separados por um espaço e colocados entre colchetes, formando uma lista. Por exemplo:

?POSY [80 45]

Se um dos valores (ou ambos) for negativo, o sinal de menos deverá estar imediatamente à esquerda do número (sem espaço). Por exemplo:

?POSXY [30 -30]

?POSXY [-50 -20]

Note que o espaço que separa o valor de X do valor de Y foi mantido. O que não pode haver é um espaço entre o número e o seu sinal.

Vendo Através de uma Janela

Até aqui, a tartaruga tem se comportado da seguinte maneira:

Se você manda avançar além do espaço limite da tela, ela simplesmente dá a volta por trás da tela, reaparecendo outra vez.

Você pode alterar esta situação digitando:

?JANELA

A partir deste instante, o que você vê na tela é uma parte do campo total da tartaruga, como se fosse uma janela. Por exemplo:

?JANELA AVANCE 500

A tartaruga anda 500 passos e não reaparece de novo na tela. Ela está longe, fora dos limites da janela. Mas você pode dar-lhe ordens que serão executadas, mesmo longe do alcance de sua visão.

Você pode limitar a ação da tartaruga apenas à tela, digitando:

?LIMITE

Se a tartaruga não estiver na janela, o Logo mostrará a mensagem:

FORA DOS LIMITES

A partir deste momento, a ação da tartaruga ficará limitada à tela. Se você digitar:

?CENTRO
?LIMITE
?AV 500

Aparecerá a mensagem:

FORA DOS LIMITES

Para voltar à condição inicial, digite:

?DEAVOLTA

Comandos vistos neste capítulo

COMANDOS	ABREV.	SIGNIFICADO
GIRE	G	Faz com que a tartaruga gire para a direita.
POSXY		Coloca a tartaruga no local X,Y
JANELA		A tela passa a ser considerada uma janela sobre o campo da tartaruga.
LIMITE		A tela passa a ser limitada. A tartaruga não pode ultrapassar os limites da tela.
DEAVOLTA		Cancela JANELA e LIMITE. A tartaruga, quando encontra os limites da tela, dá a volta por trás da tela e surge no outro extremo

OPERAÇÕES	ABREV.	SIGNIFICADO
GIRO?		Fornece o valor do ângulo formado pela tartaruga em relação ao eixo vertical central.
POS	P	Fornece as coordenadas X,Y da tartaruga.

OUTRAS FUNÇÕES

MATEMÁTICAS

12

12. OUTRAS FUNÇÕES MATEMÁTICAS

Além das operações aritméticas, a tartaruga conhece muitas outras coisas da matemática.

Operações Trigonométricas

A tartaruga é capaz de entender quase todas as operações trigonométricas. A maneira de usá-las é igual a todas as operações: deve-se associá-la a algum comando, por exemplo, **ESCREVA**.

Na tabela abaixo, vemos todas as operações trigonométricas que podem ser feitas pela tartaruga:

FUNÇÕES TRIGONOMÉTRICAS		
OPERAÇÃO	ABREVIATURA	FUNÇÃO
SENO	SEN	Calcula o seno de um ângulo dado em graus.
COSSENO	COS	Calcula o cosseno de um ângulo dado em graus.
TANGENTE n	TAN	Calcula a tangente de um ângulo dado em graus.
COTAN n	CT	Calcula a cotangente (inverso da tangente) de um ângulo dado em graus.
ASEN n		Calcula o arco (ângulo em graus) que gera um determinado valor de seno.
ACOS n		Calcula o arco (ângulo em graus) que gera um determinado valor de cosseno.

FUNÇÕES TRIGONOMÉTRICAS cont.		
OPERAÇÃO	ABREVIATURA	FUNÇÃO
ATAN n		Calcula o arco (ângulo em graus) que gera determinado valor de tangente
ACOT n		Calcula o arco (ângulo em graus) que gera determinado valor de co-tangente

n significa um valor numérico

Exemplos:

?ESCREVA SENO 30
0.5

?ESCREVA COSSENO 45
0.7071068

?ESCREVA ATAN 1
45

?ESCREVA ASEN 0.5
30

Além das funções trigonométricas, existem outras funções úteis, mostradas a seguir.

OUTRAS FUNÇÕES	
OPERAÇÃO	FUNÇÃO
INT n	<i>Fornece a parte inteira de um número, desconsiderando a parte fracionária.</i>
ARREDONDE n	<i>Fornece a parte inteira de um número, aproximando-o do número imediatamente superior, se o valor do algarismo após o ponto decimal for igual a ou maior que cinco; ou inferior, se o valor for menor que 5.</i>
RESTO n m	<i>Fornece o valor do resto de uma divisão de n por m.</i>
RAIZ n	<i>Fornece o valor da raiz quadrada de n.</i>
IGUAL? n m	<i>Fornece a palavra CORRETO, se n for igual a m. Caso contrário, a palavra ERRADO é fornecida.</i>
SORTEIE n	<i>Escolhe um valor ao acaso entre 0 e n-1.</i>

Exemplos:

?ESCREVA INT 4.7
4

?ESCREVA ARREDONDE 5.3
5

?ESCREVA ARREDONDE 8.6
8

?ESCREVA RESTO 7 3

1

?ESCREVA RAIZ 9

3

?ESCREVA IGUAL? 10 10

CORRETO

Antes de darmos um exemplo sobre o procedimento SORTEIE, vamos a algumas explicações adicionais sobre ele.

Você deve ter visto algum sorteio de prêmios pela TV. Vamos supor que se trata de um concurso com cartas enviadas pelos telespectadores. Uma delas deve ser sorteada dentre as demais. Elas então são misturadas e o apresentador apanha uma delas ao acaso.

“Ao acaso” significa que o apresentador não viu antes o que ia pegar. Poderia ser qualquer uma delas.

A operação SORTEIE faz isso: escolhe um dentre vários valores ao acaso.

Experimente:

?APRENDA LOTO

> REPITA 5 [ESCREVA SORTEIE 100]

> FIM

?LOTO

85

94

59

30

64

Note que poderiam ser quaisquer números dentre 0 e 99.

Vamos fazer agora um dado. O dado varia entre 1 e 6:

?APRENDA DADO

>ESCREVA 1 + SORTEIE 6

> FIM

Note que somamos 1 ao resultado, já que o dado não possui zero como valor e varia até 6.

Devemos ter cuidado com esta operação. Se somarmos depois, poderá ocorrer erro, porque a soma é feita antes da operação:

?ESCREVA SORTEIE 6 + 1

A linha acima é equivalente a:

?ESCREVA SORTEIE 7

O zero não é eliminado desta forma!

Devemos escrever:

?ESCREVA 1 + SORTEIE 6

Ou, então:

?ESCREVA (SORTEIE 6) + 1

Criando Operações

O Logo possui um procedimento que permite a você criar operações.

Este procedimento é SAIDA. Ele é usado da mesma forma que o primitivo ESCREVA, porém não mostra nada na tela e faz com que o procedimento tenha o comportamento de uma operação, ou seja, que se associe a outro procedimento (por exemplo ESCREVA) para poder funcionar.

Por exemplo:

```
?APRENDA PAR? :NUM  
>SAIDA 0=RESTO :NUM 2  
>FIM
```

```
?ESCREVA PAR? 10  
CORRETO
```

```
?ESCREVA PAR? 5  
FALSO
```

PALAVRAS E

LISTAS

13

13. PALAVRAS E LISTAS

Palavras, listas e números são aquilo que a tartaruga usa como dados. Por este motivo, as palavras, listas e números são chamados de “objetos”. Estamos usando os números já há alguns capítulos.

No Logo, uma palavra é um conjunto de símbolos. Os símbolos podem ser letras, algarismos ou sinais especiais. Eis algumas palavras do Logo:

OLA
X
E1234
BOLA
MONICA
#\$%@!
BONECA

Cada símbolo (também chamado de caractere) recebe o nome de elemento. BONECA é uma palavra com seis elementos, e E1234 é uma palavra com cinco elementos.

Uma palavra é separada de outros objetos por meio de espaços ou de um dos seguintes sinais:

*[] () < > + - * / (colocados depois da palavra) ” e : (colocados imediatamente antes da palavra, ou seja, sem espaço entre o sinal e a palavra)*

Quando se colocam dois pontos antes da palavra, ela passa a ser considerada como o nome de uma variável.

Quando se usam as aspas sozinhas, isto é, não acompanhadas de algum elemento que as siga, indicam uma “palavra vazia”.

Quando colocamos um valor numérico depois das aspas, o número passa a ser considerado uma palavra.

Os sinais separadores poderão ser usados como elementos de uma palavra, se forem colocados depois da barra invertida (\). A barra invertida é obtida por SYMBOL D. Por exemplo:

?ESCREVA “COLCHETES\[\]
COLCHETES[]

A barra invertida também pode ser usada para se incluírem espaços em branco numa palavra:

?ESCREVA "BOM\ DIA
BOM DIA

Já falamos sobre as listas anteriormente. Vimos que uma lista é um conjunto de objetos Logo colocados entre colchetes. Cada objeto dentro de um a lista é chamado de "item".

Os itens podem ser: palavras, listas ou listas de listas. Se uma lista não contém itens ela é chamada de "lista vazia".

O Logo possui diversos comandos e operações destinados à manipulação de palavras e listas. A operação SEMPRIMEIRO (abrevia-se SP) fornece uma palavra sem o primeiro elemento ou uma lista sem o primeiro item. Por exemplo:

?ESCREVA SEMPRIMEIRO [CHAPÉUZINHO VERMELHO]
VERMELHO

?ESCREVA SEMPRIMEIRO "LOBA
OBA

Do mesmo modo, a operação SEMULTIMO (abrevia-se SU) permite-nos escrever uma lista sem o último item ou uma palavra sem o último elemento. Por exemplo:

?ESCREVA SEMULTIMO [LOBO MAU]
LOBO

?ESCREVA SEMULTIMO "CASAS
CASA

A operação VAZIO? permite que a tartaruga verifique se o conteúdo de um objeto é ou não vazio. Se for vazio, a tartaruga dirá CORRETO. Se não for vazio, a tartaruga dirá FALSO.

Exemplos:

?ATRIBUA "CAIXA "TESOURA
?ATRIBUA "A []
?ESCREVA VAZIO? :A
CORRETO

?ESCREVA VAZIO? :CAIXA
FALSO

A operação IGUAL?, descrita no capítulo anterior, inclui também listas e

palavras. Ela fornece a palavra **CORRETO**, como resposta, se os **objetos** forem iguais. Por exemplo:

?ESCREVA IGUAL? "PALAVRA SEMPRIMEIRO [UMA PALAVRA]
CORRETO

?ESCREVA IGUAL? 10 23/5
FALSO

?ESCREVA IGUAL? [LISTA PALAVRA] [LISTA PALAVRA]
CORRETO

A operação **CONTE** dá o número de elementos de uma palavra ou o número de itens de uma lista. Por exemplo:

?ESCREVA CONTE [A B C D E F]
6

?ESCREVA CONTE "QWERT
5

A operação **PRIMEIRO** dá o primeiro elemento de uma lista ou palavra. Por exemplo:

?ESCREVA PRIMEIRO [HOJE FOI UM DIA CHUVOSO]
HOJE

?ESCREVA PRIMEIRO "LOBO
L

A operação **ULTIMO** dá o último elemento de uma palavra ou o último item de uma lista. Por exemplo:

?ESCREVA ULTIMO [HOJE FOI UM DIA CHUVOSO]
CHUVOSO

?ESCREVA ULTIMO "LOBO
O

Para pedir à tartaruga qualquer item de uma lista, usase a operação **ITEM**. Deve-se digitar **ITEM** seguido pelo número do item desejado:

?ESCREVA ITEM 5 [LOBO MAU E OS TRES PORQUINHOS]
TRES

Para que a tartaruga inclua um item numa lista, podemos usar as operações **NOINICIO** e **NOFIM**.

NOINICIO coloca um item no começo de uma lista ou um elemento no começo de uma lista. Por exemplo:

```
?ESCREVA NOINICIO "LOBO [MAU E OS TRES PORQUINHOS]
LOBO MAU E OS TRES PORQUINHOS
```

NOFIM coloca um item no final de um lista ou um elemento no final de uma lista. Por exemplo:

```
?ESCREVA NOFIM "GORDOS [OS TRES PORQUINHOS]
OS TRES PORQUINHOS GORDOS
```

A operação *LISTA* contrói uma lista a partir dos objetos fornecidos. Por exemplo:

```
?ATRIBUA "A LISTA "LOBO "MAU
?ATRIBUA "B LISTA [LOBO MAU] [TRES PORQUINHOS]
```

A variável *A* recebe a lista *[LOBO MAU]*, e a variável *B* recebe a lista *[[LOBO MAU] [TRES PORQUINHOS]]*.

Se o número de objetos for maior que 2, deve-se escrever a operação entre parênteses. Por exemplo:

```
?ATRIBUA "C (LISTA [LOBO MAU] [TRES PORQUINHOS]
[CHAPÉUZINHO VERMELHO])
```

A variável *C* recebe a lista *[[LOBO MAU] [TRES PORQUINHOS] [CHAPÉUZINHO VERMELHO]]*.

Para verificar se um dado objeto é uma lista, existe a operação *LISTA?*, que pergunta à tartaruga esta condição. Por exemplo:

```
?ATRIBUA "A "LOBO\ MAU
?ESCREVA LISTA? :A
FALSO
```

```
?ATRIBUA "A [LOBO MAU]
?ESCREVA LISTA? :A
CORRETO
```

Podemos também perguntar se uma determinada palavra ou lista é um item de uma lista. Usa-se a operação *MEMBRO?* Por exemplo:

```
?ESCREVA MEMBRO? "TESOURA [COLA LAPIS TESOURA
PAPEL]
CORRETO
```

?ESCREVA MEMBRO? "BORRACHA [COLA LAPIS TESOURA PAPEL]
FALSO

?ESCREVA MEMBRO? [LOBO MAU] [[LOBO MAU] [TRES PORQUINHOS]]
CORRETO

?ESCREVA MEMBRO? [CHAPEUZINHO VERMELHO] [[LOBO MAU] [TRES PORQUINHOS]]
FALSO

Se quisermos saber se um objeto é um número ou não, podemos usar a pergunta VALOR?.

?ESCREVA VALOR? 1
CORRETO

?ESCREVA VALOR? "LETRAS
FALSO

?ESCREVA VALOR? "1
FALSO

?ESCREVA VALOR? [1]
FALSO

Note estes dois últimos exemplos. A tartaruga não considera como número um valor colocado depois das aspas, também não considera como número um valor escrito como uma lista.

Podemos ainda verificar se um dado objeto é ou não uma palavra. A pergunta usada é PALAVRA?.

?ESCREVA PALAVRA? "HOTEL
CORRETO

?ESCREVA PALAVRA? "1000
CORRETO

?ESCREVA PALAVRA? 1000
FALSO

?ESCREVA PALAVRA? [LOBO MAU]
FALSO

?ESCREVA PALAVRA? "
CORRETO

Podemos usar uma operação que constrói palavras. Esta operação reúne, numa única palavra, duas ou mais palavras. Se forem usados mais de dois objetos, colocam-se parênteses nesta operação. Por exemplo:

?ESCREVA PALAVRA "FO "GUE "TE
FOGUETE

Da mesma forma que podemos contruir uma palavra podemos construir uma sentença, através da operação FRASE (abrevia-se FR).

A operação FRASE reúne dois ou mais objetos, formando uma lista. Se o número de objetos for maior do que 2, deve-se usar parênteses.

Por exemplo:

?ESCREVA (FRASE "ATRAVES "DO "ESPELHO "E "O "QUE
"ALICE "ENCONTROU "LA)

ATRAVES DO ESPELHO E O QUE ALICE ENCONTROU LA

?ESCREVA (FRASE "O [LOBO MAU] "PERSEGUE "OS [TRES
PORQUINHOS])

O LOBO MAU PERSEGUE OS TRES PORQUINHOS

Note a diferença desta operação com a operação LISTA. LISTA considera as listas como listas. FRASE considera todos os itens como palavras, quer sejam palavras ou listas.

Uma lista pode conter uma linha de comandos do Logo. Enquanto estiverem na lista, os comandos não serão executados, pois a tartaruga considera-os apenas como objetos.

Entretanto existe um comando que permite à tartaruga considerar uma lista com comandos como uma linha executável. Trata-se do comando EXECUTE. Por exemplo:

?EXECUTE [ESCREVA [BOM DIA]]
BOM DIA

O Código ASCII

Na realidade, o computador só pode entender números. para que ele possa

entender palavras, letras e símbolos, estes devem ser codificados. Existe um código padrão, chamado código ASCII.

Uma tabela do código ASCII pode ser vista no apêndice D do manual de seu computador. A primeira coluna desta tabela mostra o código, e a segunda, o símbolo correspondente. A partir do valor 32 até 127 são as letras, algarismos e símbolos usados no Logo.

Existem duas operações no Logo que trabalham diretamente com este código. A primeira delas é a operação ASCII que fornece o código de um símbolo. Por exemplo:

```
?ESCREVA ASCII "A  
65
```

```
?ESCREVA ASCII "B  
66
```

(O símbolo pesquisado deve vir sempre depois de “)

A outra operação é SIMBOLO, que faz justamente o contrário: fornece o símbolo que possui um determinado código ASCII, situado entre 32 e 127.

```
?ESCREVA SIMBOLO 65  
A
```

```
?ESCREVA SIMBOLO 66  
B
```

Nomes de Variáveis

Qualquer palavra em Logo pode ser usada como nome de uma variável. Como podemos então saber quando uma palavra foi usada para nomear uma variável? Podemos perguntar à tartaruga por meio de VAR?. Se a palavra for o nome de uma variável, a tartaruga responderá CORRETO. Se a Palavra não for o nome de uma variável, a tartaruga responderá FALSO.

Por exemplo:

```
?ATRIBUA "HISTORIA "CHAPEUZINHO\ VERMELHO  
?ESCREVA VAR? "HISTORIA  
CORRETO  
?ESCREVA VAR? "CHAPEUZINHO  
FALSO
```

(se CHAPEUZINHO não tiver sido usada como variável)

Uma outra coisa interessante a respeito dos nomes de variáveis é que eles podem ser usados como palavras, numa frase ou lista, ou como conteúdo de outras variáveis. Por exemplo:

```
?ATRIBUA "LOBO "MAU
?ATRIBUA "MAU "VILAO
```

```
?ATRIBUA "HISTORIA [CHAPEUZINHO VERMELHO]
?ESCREVA [A HISTORIA DO LOBO MAU]
A HISTORIA DO LOBO MAU
```

Existe em Logo a operação OBJETO. Ela fornece o conteúdo de um nome de uma variável, mesmo que esta variável seja uma palavra armazenada em outra variável. Por exemplo:

```
?ATRIBUA "MARIA "FELIZ
?ATRIBUA "FELIZ "ANIVERSARIO
```

```
?ESCREVA OBJETO "MARIA
FELIZ
```

```
?ESCREVA OBJETO :MARIA
ANIVERSARIO
```

Observe que, quando MARIA é tratada como uma palavra (colocada depois de "), a operação OBJETO fornece seu conteúdo. Se for tratada como uma variável (colocada depois de :), OBJETO fornece o conteúdo do nome que nela está armazenado.

Verifique agora um exemplo numérico:

```
?APRENDA INC :X
>ATRIBUA :X 1 + OBJETO :X
>FIM
```

```
?ATRIBUA "TOTAL 7
?ESCREVA :TOTAL
7
```

```
?INC "TOTAL
?ES :TOTAL
8
```

O procedimento INC usou o conteúdo do nome TOTAL para fazer o cálculo.

Você pode mudar o conteúdo de um nome da mesma forma que altera um procedimento, por meio do EDITOR.

Existe um comando Logo que permite levar todos os nomes de variáveis e seu conteúdo para o EDITOR. Assim podemos alterá-los. Este comando é EDNOME.

Quando você usa EDNOME, a tartaruga dá todas as variáveis obtidas com o comando ATRIBUA e seu último valor. Por exemplo:

?EDNOME

ATRIBUA "A 10

ATRIBUA "NOME "PALAVRAS

ATRIBUA "HISTORIA [CHAPEUZINHO VERMELHO]

Depois de colocado no EDITOR, cada nome poderá ser modificado e seu conteúdo alterado. Para isso, você usa os mesmos comando de edição dos procedimentos.

Comandos e Operações vistos neste capítulo

OPERAÇÃO	ABREV.	SIGNIFICADO
SEMPRIMEIRO	SP	Mostra uma palavra sem a primeira letra ou uma lista sem o primeiro elemento.
SEMULTIMO	SV	Mostra uma palavra sem a ultima letra ou uma lista sem o ultimo elemento.
IGUAL?		Verifica se duas palavras dois valores ou duas lista são iguais.
PRIMEIRO		Mostra a primeira letra de uma palavra ou o primeiro elemento de uma lista.
ULTIMO		Mostra a ultima letra de uma palavra ou o ultimo elemento de uma lista.
CONTE		Conta os elementos de uma lista ou as letras de uma palavra.
ITEM		Mostra o enézimo item de uma lista.
NOINICIO		Coloca um elemento no início da lista.
NOFIM		Coloca um elemento no fim da lista.
LISTA		Compõe uma lista com os elementos indicados.

<i>OPERAÇÃO</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
LISTA?		<i>Verifica se um determinado conjunto de elementos forma uma LISTA.</i>
MEMBRO?		<i>Verifica se um elemento é membro de uma lista.</i>
VALOR?		<i>Verifica se um determinado elemento é um valor numérico.</i>
PALAVRA		<i>Transforma um determinado conjunto de letras ou sílabas numa palavra.</i>
PALAVRA?		<i>Verifica se um determinado dado é uma palavra.</i>
FRASE	FR	<i>Transforma um conjunto de dados numa frase.</i>
ASCII		<i>Verifica o código do primeiro caractere de uma palavra.</i>
VAR?		<i>Verifica se uma determinada palavra é o nome de uma variável.</i>
OBJETO		<i>Mostra o conteúdo de um nome.</i>

<i>COMANDOS</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
EXECUTE		<i>Execute os procedimentos contidos numa lista.</i>
EDNOME		<i>Permite a edição do conteúdo das variáveis obtidas com o procedimento ATRIBUA.</i>

ENTRADAS

E SAIDAS

14

14. ENTRADAS E SAÍDAS

Este capítulo contém os comandos primitivos, que permitem a você comunicar-se com a tartaruga por meio do teclado e receber as respostas da tartaruga por meio da tela (ou por meio do alto-falante).

Você já conhece um destes comandos: ESCREVA. Este comando exibe na tela uma palavra ou o conteúdo de um lista.

Existe um outro comando com função semelhante, o comando MOSTRE. Ele mostra na tela uma palavra ou uma lista. Quando se tratar de uma lista, MOSTRE colocará na tela os colchetes. Por exemplo:

?MOSTRE "TEXTO
TEXTO

?MOSTRE [CHAPÉUZINHO VERMELHO]
[CHAPÉUZINHO VERMELHO]

Existe um terceiro comando para mostrar textos na tela, o DIGITE. DIGITE funciona da mesma forma que o comando ESCREVA, só que o comando ESCREVA manda a tartaruga escrever, na próxima linha, o próximo texto a ser escrito, enquanto que DIGITE manda a tartaruga continuar na mesma linha.

?ESCREVA "BOM ESCREVA "DIA
BOM
DIA

?DIGITE "BOM DIGITE "DIA
BOMDIA

Além desses comandos, o Logo possui as seguintes operações:

LEIATECLA (abrevia-se LT). Esta operação permite que você digite uma tecla ou combinação de teclas, sem apresentar os caracteres imediatamente na tela.

TECLA?. Esta operação verifica se uma tecla ou combinação de teclas foi digitada. Se alguma tecla foi digitada, a tartaruga responde CORRETO.

Os procedimentos a seguir exemplificam as duas operações descritas.

A tartaruga anda para a frente se uma tecla é digitada e vira 30 graus à direita, se a tecla P for pressionada, ou 30 graus à esquerda se a tecla Q for pressionada:

?APRENDA VIRAR

> AV 2

> SE TECLA? [RODE LEIA TECLA]

> VIRAR

> FIM

?APRENDA RODE :DIR

> SE :DIR="P [DT 30]

> SE :DIR="Q [EQ 30]

> FIM

A operação LEIALISTA (abrevia-se LL) permite que se dê entrada a uma lista. A tartaruga aguardará a digitação de uma frase (terminada por <ENTER>). Exemplo:

?APRENDA HISTORIA.SEM.FIM

> ESCREVA [ESPELHO, ESPELHO MEU, EXISTE ALGUÉM MAIS BELA DO QUE EU?]

> SE LEIALISTA= [NAO, MAJESTADE, VOCE E' A MAIS BELA] [ES [A RAINHA SORRI] PARE] [ES [A RAINHA ZANGA-SE E PERGUNTA NOVAMENTE:] HISTORIA.SEM.FIM]

> FIM

Neste procedimento, a tartaruga mostra a frase inicial e espera que você digite a resposta do espelho. Se a resposta for NAO MAJESTADE, VOCE E' A MAIS BELA, a rainha sorri e o procedimento pára. Em caso contrário, ela perguntará a você até receber uma resposta satisfatória.

Impressora

Para copiar os desenhos da tela pela impressora, você deve usar o comando IMPTELA.

Para copiar tudo que aparece na tela (texto e desenhos) pela impressora, você deve usar o comando IMPLIG.

IMPDES

Para voltar à condição inicial, use IMPDES.

Robô

Através do comando LIGAROBO os comandos da tartaruga são transferidos a um robô que está conectado à saída EXPANSION do computador. Se não houver nenhum robô conectado, pode ocorrer a perda de dados e do programa Logo. Apenas use este comando se tiver o robô conectado ao computador.

Com o comando *DESLROBO*, desliga-se o robô, transferindo os comando de volta à tartaruga da tela.

Ouvindo a Tartaruga

A tartaruga do Logo não é muda. Ela pode emitir sons, se você usar o comando *SOM*.

O comando *SOM* deve ser seguido por uma lista com dois valores. O primeiro deles é relativo à duração do som produzido, e o segundo é relativo à frequência. O primeiro valor varia entre 0 e 255, e o segundo varia de -62 a 75.

Exemplo:

```
?SOM [1 0]

?APRENDA TOCAR :A :B
>SE :A>255 OU :A<0 [PARE]
>SE :B<-62 OU :B>255 [PARE]
>SOM [:A :B]
>TOCAR :A+1 :B+1
```

Se você quiser que a tartaruga espere durante algum tempo antes de executar suas ordens, pode usar o comando *ESPERE*. Este comando deve ser seguido do tempo que você quer que a tartaruga espere. *Exemplo:*

```
?APRENDA CAMARALENTA :TEMPO
>REPITA 80 [AV 1 ESPERE :TEMPO]
>FIM
```

Comandos e operações vistos neste capítulo.

<i>COMANDO</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
MOSTRE		<i>Equivale a ESCREVA porém, mostra os colchetes das listas.</i>
DIGITE		<i>Equivale a ESCREVA, porém se forem usados dois comandos na mesma linha, as mensagens serão impressas na mesma linha</i>
IMPTELA		<i>Copia a tela na impressora.</i>
IMPLIG		<i>Liga impressora.</i>
IMPDES		<i>Desliga impressora.</i>
LIGAROBO		<i>Liga robo conectado à saída de expressão.</i>
PESLROBO		<i>Desliga o robo.</i>
SOM		<i>Produz um som no alto-falante da TV.</i>

<i>OPERAÇÃO</i>	<i>ABREV.</i>	<i>SIGNIFICADO</i>
LEIATELA?	LT	<i>Permite a introdução de caractere, como dados o computador aguarda que uma pressionada</i>
TECLA?		<i>Verifica se uma tecla foi pressionada</i>
LEIALISTA		<i>Permite a introdução de uma lista como dado. (O computador aguarda que seja pressionado <ENTER>)</i>

OUTROS COMANDOS E OPERAÇÕES

15

15. OUTROS COMANDOS E OPERAÇÕES

Além dos comandos e operações vistos nos capítulos anteriores, o Logo ainda possui os seguintes:

Comandos de Tela

AJUSTEXY

Comando que estabelece a relação entre o número de passos da tartaruga no eixo X e no eixo Y (original 50/60). Deve ser seguido de uma lista de dois valores [X Y].

Exemplo:

```
?AJUSTEXY [40 60]
```

RELXY?

Operação que fornece a relação de x/y. No exemplo anterior, teríamos:

```
ES RELXY  
40 60
```

PONTO

Comando que coloca um ponto na tela, nas coordenadas X e Y. A tartaruga não se move. Deve ser seguido de uma lista de dois valores [X Y].

DIREÇÃO

Operação que fornece o valor do giro da tartaruga, para que esta fique apontando para determinado ponto de coordenadas X e Y. Deve ser seguido de uma lista de dois valores [X Y]. Exemplo:

```
?ES DIREÇÃO [10 10]  
45
```

Espaço de trabalho

APTUDO

Comando que apaga tudo o que estiver na memória da tartaruga.

APVAR

Somando que apaga o conteúdo de uma variável. Deve ser seguido de: "nome da variável". Exemplo:

?APVAR "A

APTVAR

Comando que apaga todas as variáveis da memória da tartaruga.

APTP

Comando que apaga todos os procedimentos que a tartaruga aprendeu.

APARQ

Comando que apaga arquivo em unidade de disco.

CATALOGO

Comando que mostra na tela o conteúdo de unidade de disco.

DRIVE

Comando que permite escolher a unidade de disco a ser usada. Deve ser seguido de um número de 0 a 8. Quando é seguido de 0, volta à condição normal do Logo.

Definições e Redefinições

COPIEDEF

Comando que copia as definições de um procedimento em outro. Deve ser seguido de: "Nome" novo nome. Exemplo:

?COPIEDEF "QUADRADO "Q

Com este comando, passamos a ter, além do procedimento QUADRADO, o procedimento Q. Este último faz o mesmo que o primeiro.

DEFINA

Comando que transforma uma lista numa definição de procedimento. Por exemplo:

?DEFINA "TESTE [[:X :Y] [ESCREVA :X] [ESCREVA :Y]]

É equivalente a:

```
?APRENDA TESTE :X :Y
>ESCREVA :X
>ESCREVA :Y
>FIM
```

DEF?

Operação que fornece CORRETO, se o nome de que é seguida for um procedimento. Em caso negativo, fornece FALSO. Exemplo:

```
?ES DEF? "QUADRADO
CORRETO
```

PRIMITIVO?

Operação que fornece CORRETO, se o nome de que é seguida for um procedimento primitivo. Exemplos:

```
?ESCREVA PRIMITIVO? "AVANCE
CORRETO
```

```
?ESCREVA PRIMITIVO? "QUADRADO
FALSO
```

TEXTO

Operação que fornece a definição de um procedimento como um lista. Exemplo:

```
?ES TEXTO "TESTE
[:X :Y] [ESCREVA :X] [ESCREVA :Y]
```

Funções Diversas

NIVEL1

Comando que pára o procedimento, fazendo o Logo retornar ao nível inicial. O sinal > volta a ser ?. Este comando é fiferente de PARE, porque se PARE estiver em um procedimento chamado por outro, ele retorna ao procedimento que fez a chamada.

TCHAU

Comando que faz o computador abandonar o Logo e entrar no BASIC.

NOS?

Operação que fornece o número de nós disponível na memória. Um nó possui 5 bytes.

ARRUME

Comando que reorganiza o espaço livre da memória, eliminando “sujeira” deixada pelas constantes alterações de valores de variáveis, que diminuem o espaço livre para uso do LOGO.

.CONTEM?

Operação que mostra o buffer, indicando as últimas coisas digitadas. O ponto que antecipa o nome deve ser digitado.

.RESERVE

Reserva uma área da memória para armazenar um arquivo binário (tipo BIN) ou um programa em linguagem de máquina. O comando deve ser seguido pelo número de bytes necessários para os dados ou programas. Por exemplo.

? .RESERVE 500

Não esqueça do ponto em qualquer comando que o utilize.

.RESERVADO

A operação .RESERVADO fornece os endereços inicial e final da área reservada. Por exemplo:

**? ES RESERVADO
64974 65024**

.GRAVEB

Este comando permite a gravação de um arquivo binário (dados ou programas em linguagem de máquina). O comando deve ser seguido por um nome de arquivo e uma lista com o endereço que marca o início do arquivo e o seu tamanho em bytes.

O arquivo binário é o conteúdo da área reservada por .RESERVE, de acordo com o seguinte mapa:

RAM	Variáveis do sistema LOGO	65535
	.RESERVE ENDEREÇO INICIAL	65024
	Memória de trabalho	57770
	Interpretador	24832
	Variáveis do sistema BASIC	16384
ROM	BASIC	000

Por exemplo:

?GRAVEB "BINARIO [64974 500]

.LEIAB

Permite a leitura de um arquivo binário (seguido pelas letras BIN na apresentação do nome do arquivo). Deve ser especificado o endereço inicial do arquivo. Por exemplo:

?LEIAB "BINARIO 64974

.COLOQUE

Comando que permite a colocação de um valor em um determinado endereço. Deve ser indicado primeiro o endereço, seguido pelo valor. Evite fazer isso na área reservada á variáveis do sistema (tanto do BASIC como do LOGO) ou na área do interpretador LOGO. Use apenas a área do espaço de trabalho ou a área reservada pelo comando .RESERVE. Exemplo:

?COLOQUE 65000 10

.EXAMINE

Operação que permite verificar o conteúdo de um endereço. Por exemplo.

?ES .EXAMINE 65000
10

.CHAME

Comando que chama uma rotina em linguagem de máquina, armazenada num determinado endereço. Por exemplo.

`?CHAME 65000`

.VELTRANS

Comando que indica a velocidade de transmissão para comunicação de dados serial. Deve ser indicado o valor em Baud rate desta velocidade. Estes valores podem ser:

50, 110, 300, 600, 1200, 2400, 4800 ou 19200

.ENTSER

Comando que lê o que for enviado através de uma interface RS232

.SAISER

Comando que permite enviar-se um byte através de uma interface RS232. Por exemplo:

`.SAISER 10`

O valor enviado deve estar entre 0 e 255

Composição, Fitolito e Impressão:
BANDEIRANTE S.A. GRÁFICA E EDITORA
Fone: 452-3444

